

# curso de música

TEORIA E  
PRÁTICA

**MSX**



**Editora  
Aleph**

**Barbieri - Piazzzi**

UPE 105 2012-2013.40  $\phi 2$

# curso de música

vol.1



**Barbieri - Piazzì**

# **curso de música**

**vol.1**



© EDITORA ALEPH 1988

Coordenação Editorial : RENATO DA SILVA OLIVEIRA  
Coordenação Pedagógica : BETTY FROMER PIAZZI  
Produção Editorial : ROSA KOGAN FROMER  
Digitação : ROSA MARIA X.M. SHIGENO  
Ilustrações : DURVALY ODILON NICOLETTI  
Arte e Capa : ANA LÚCIA ANTICO



ALEPH PUBLICAÇÕES E  
ASSESSORIA PEDAGÓGICA LTDA.  
Caixa Postal: 20707 - CEP: 01498  
São Paulo-SP Tel.:(011) 843-3202

**Dados de Catalogação na Publicação (CIP) Internacional  
(Câmara Brasileira do Livro, SP, Brasil)**

Barbieri, Eduardo Lamounier, 1956-  
8191c Corno de música / Barbieri-Piazzzi. -- São Paulo :  
v.1-2 Aleph, 1988.  
(Série didática)

1. MSX (Computadores) 2. Música - Ensino auxiliado  
por computador 3. Música - Estudo e ensino I. Piazzzi,  
Pierluigi, 1943- II. Título. III. Série.

88-0479

CDD-780.702854  
-001.66

**Índices para catálogo sistemático:**

1. Computadores : Ensino de música 780.702854
2. MSX : Computadores : Processamento de dados 001.66
3. Música : Ensino por computador 780.702854
4. Música eletrônica 780.702854



## SUMÁRIO

NOTA DO EDITOR .....	6
AULA 1 - A MÚSICA E O MSX .....	9
AULA 2 - A NOTAÇÃO MUSICAL PARA A ALTURA .....	17
AULA 3 - A NOTAÇÃO MUSICAL PARA A DURAÇÃO .....	31
AULA 4 - A DIVISÃO DA MÚSICA .....	47
AULA 5 - AS ESCALAS MUSICAIS .....	71
AULA 6 - OS ACIDENTES MUSICAIS .....	89

### APÊNDICES

A - EXERCÍCIOS SUPLEMENTARES .....	109
B - RESPOSTAS DOS EXERCÍCIOS .....	115
C - O COMANDO PLAY .....	133
D - BIBLIOGRAFIA RECOMENDADA .....	141



## NOTA DO EDITOR

Quando eu tinha cinco anos fui enviado a uma professora de piano para aprender os rudimentos musicais. Ao invés de ser apresentado ao maravilhoso mundo da música através de um processo de descoberta do som, da melodia e da harmonia, fui massacrado por enfadonhas horas de solfejo e leitura de partituras.

As poucas melodias que me permitiam dedilhar não me diziam absolutamente nada! O que estavam fazendo comigo não tinha nada a ver com música, na medida que não me proporcionava o menor prazer.

Para completar a tortura, durante o solfejo a professora mantinha um lápis muito bem apontado pendurado sobre minha cabeça. Se eu errasse na leitura, ela soltava o lápis que acertava dolorosamente em meu crânio!

Essa versão "didática" da espada de Dâmocles me traumatizou a tal ponto que criei uma "tosse nervosa" como truque para ganhar tempo na leitura da partitura. O médico ao qual fui levado me curou num instante mandando suspender imediatamente as aulas de piano!

Essa minha história, com algumas variações, é a de milhares de "músicos frustrados" que por não terem tido uma iniciação adequada, acabaram se afastando do mundo maravilhoso da música tornando-se, quando muito, "ouvintes passivos".

Obviamente nem todos tiveram o privilégio que eu tive de se defrontar com uma professora que, com certeza, fez um curso de aperfeiçoamento num conservatório da GESTAPO. A dificuldade, porém, que todos nós

temos de treinar uma psicometricidade adequada para tocar algum instrumento musical, já representa um obstáculo suficiente para desanimar a maioria das pessoas.

Só depois dos quarenta é que descobri o instrumento musical adequado para estudar música de maneira agradável: meu microcomputador! Nele eu posso digitar partituras, fazer arranjos, experimentar novos acordes, timbres e ritmos, sem precisar desenvolver uma habilidade manual que me tomaria horas e horas de estudo enfadonho.

Esta descoberta foi tão gratificante que resolvi compartilhá-la com todos os outros "músicos desajeitados", marginalizados do mundo musical.

Para isso convidei o EDUARDO LAMOUNIER BARBIERI, um jovem engenheiro eletrônico, brilhante músico e compositor, familiarizado tanto com a Música quanto com os meios eletrônicos de produzi-la.

Da minha colaboração com o BARBIERI surgiu este CURSO DE MÚSICA, dividido em dois volumes.

No primeiro, ensinamos os primeiros passos da teoria musical, utilizando essencialmente a macro-linguagem do comando PLAY do MSX. Nele discutimos as qualidades "fisiológicas" do som e o surgimento das escalas musicais sob uma perspectiva histórica.

Tentamos utilizar os exemplos mais ecléticos possíveis, indo desde cantigas infantis e chegando até o jazz, passando pela música medieval e clássica.

O nosso objetivo final, neste volume, é levar o leitor a fazer a transcrição de partituras até certo ponto complexas, com conhecimento não só da notação musical mas também com uma boa noção da teoria musical que elas envolvem.

No segundo volume, além de completar as noções de teoria musical iniciadas no primeiro volume, discutimos as qualidades "físicas" do som, utilizando o comando SOUND do MSX e mostrando a "construção" de tons e ruídos no computador com a matematização de determinados parâmetros.

Em nenhum dos dois volumes foi nossa intenção ensinar programação, apesar deste aprendizado acabar ocorrendo como sub-produto do curso.

O principal objetivo do CURSO DE MÚSICA foi o de realmente promover a descoberta do mundo musical em si. Além de pensar no "músico frustrado" que citei anteriormente, pensamos no jovem que deseja se iniciar passando por uma cultura que já lhe é relativamente familiar: a micro-informática.

Este livro, a rigor, não foi escrito a quatro mãos, como diriam os pianistas, mas sim a 12 dedos!

Enquanto o BARBIERI utilizou plenamente suas duas mãos de tecladista experiente e músico de "mão cheia" (com o perdão do trocadilho), eu fui dedilhando num cantinho do teclado usando apenas os dois dedos de músico principiante, numa espécie de "bife" editorial complementando o texto do BARBIERI com programas e exercícios.

Para um bom aproveitamento do CURSO DE MÚSICA é aconselhável, mas não indispensável, que o leitor tenha algum conhecimento do BASIC MSX, o que pode ser conseguido com uma leitura dos manuais que acompanham o micro ou estudando a bibliografia aconselhada no fim de cada volume.

Esperamos que esta experiência editorial inovadora seja de real ajuda aos nossos leitores e aguardamos críticas e sugestões que nos permitam aprimorar e complementar este trabalho que tira o microcomputador da exclusiva tarefa de atividades profissionais (ou de "joguinhos") e o coloca como um valioso instrumento para ingressar no mundo das artes.

Pierluigi Piazzì  
Marco de 1988





## INTRODUÇÃO HISTÓRICA

Música, palavra mágica que despertou a atenção de artistas, sábios, filósofos, governantes, e que os levou ao sorriso ou às lágrimas. Mas o que significa essa palavra? Vamos voltar um pouco no tempo... A palavra "música" tem origens na Grécia Antiga, onde **MOUSIKE** representava não apenas o som dos instrumentos e o canto mas a arte das musas: os sons, os cantos, a poesia, a dança, enfim, tudo o que pertencia às musas. Uma época em que o conjunto de lendas que constituíam sua mitologia era também o que servia de explicação para os mistérios do mundo. Os poetas da época de Homero, por exemplo, utilizavam-se de instrumentos como a cítara para acompanhar sua poesia, o que evidência uma grande relação entre a música, a poesia e o teatro. Na civilização grega, a incapacidade para a arte e para a música, em particular, era motivo de frustração e infelicidade.

Há muito o que se aprender com a Grécia Antiga. Ela foi o palco de um grande desenvolvimento da música, seja pela concepção e aprimoramento de instrumentos musicais como a flauta, a lira, a trombeta, a cítara, a gaita de foles, uma infinidade de instrumentos de percussão, seja pelo canto coral, seja pelas teorias e relações numéricas desenvolvidas por Pitágoras, principalmente. Talvez o mais antigo documento escrito sobre a música grega seja da tragédia Orestes, de Eurípedes, datado do século V a.C., embora

os estudos para representação da música grega tenham se iniciado ao redor do século VII a.C., ou seja, há mais ou menos 2.600 anos.

Contudo, se compararmos com os estudos feitos pelo sábio chinês Ling Lun, que ao redor de 2500 a.C. estabeleceu regras para o sistema pentatônico e com os instrumentos musicais do Ceilão, ao redor de 7000 a.C. veremos que há 10000 anos a música vem se desenvolvendo e atraindo o interesse de alguns pesquisadores.

Figura 1.1 - Detalhe de um antigo vaso grego com Peléu tocando cítara.



Pitágoras, Homero, Aristóximo, Euclides, Ptolomeu, todos têm uma ligação muito grande...a música. Este é o elo que une você a todos esses gênios...

## ALGUNS CONCEITOS BÁSICOS

Quando ouvimos um pássaro cantar, um gato miar ou uma criança ensaiar seus primeiros gorgolejos, percebemos que esses sons, dependendo do nosso humor, podem evocar sentimentos bastante variados, da alegria à tristeza.

O homem, devido à sua capacidade de raciocínio, aliada à lógica, tem criado padrões para a utilização, geração e execução dos sons de forma a tornar mais agradável a transmissão desses sentimentos. Esta combinação, que para o homem é lógica, é chamada de música.

O conceito de música é, portanto, bastante abrangente. Contudo, há várias tendências e características no agrupamento dos sons, que tornam a definição de música muito pessoal e também dependente da época que analisamos.

O fato é que tudo o que impressiona os órgãos auditivos, dependendo do nosso estado de espírito, pode aguçá-lo ou refrear nossos sentimentos e transformar nossas atitudes. Basta lembrar o efeito das marchas militares para infundir coragem antes da batalha ou corais religiosos para criar um clima de espiritualidade e meditação. Assim, a música contribui de uma forma bastante significativa quando aparece sob a forma de efeitos sonoros num filme, como estímulo para a dança, ou como forma de relaxamento, na musicoterapia.

Estas expressões, como inúmeras outras, representam os atuais modos de utilizar a música com finalidades específicas.

## OS INSTRUMENTOS MUSICAIS

Para uma dada utilização da música, utilizam-se instrumentos geradores de som. Temos, assim, um novo conceito: um instrumento musical. Todo instrumento gerador de som pode se tornar um instrumento musical.

A utilização de um tipo de instrumento musical pode refletir o avanço tecnológico de uma determinada época bem como ajudar a compreendê-la melhor.

Ao redor do ano 4000 a.C. (há quase seis mil anos), os egípcios batiam paus e pedras para acompanhar seu canto. Se pudéssemos viajar pelo tempo, veríamos na Palestina, no século X a.C., trompetes, instrumentos de corda e corais no templo de Salomão. Ou ainda Nero assistindo a um recital de hydraulis, um tipo de órgão de tubos.

Atualmente, além dos instrumentos tradicionais o desenvolvimento tecnológico colocou à nossa disposição uma grande quantidade de instrumentos eletrônicos como órgãos, sintetizadores, emuladores e até o próprio microcomputador que é o instrumento básico utilizado neste livro.

Paralelamente à evolução tecnológica dos instrumentos musicais ocorreram grandes avanços no estudo da representação da música como linguagem. O sistema de notação musical atual obedece a certas regras, como qualquer outra linguagem, mas permite a inserção de uma série de novos parâmetros, devido à complexidade de formas e de sons de que atualmente dispomos.

Esta característica é muito importante pois alguns instrumentos musicais têm hoje dispositivos de automação, ou seja, podem ser programados para tocar, sem a necessidade de que o programador passe anos e anos se educando e exercitando para adquirir rapidez, força e sensibilidade para poder tocá-lo.

Seja qual for o tipo de instrumento que você decida aprender a tocar, é necessário descobrir como funciona e quais seus recursos, para poder aproveitá-lo ao máximo. A pesquisa na geração e combinação de sons para produzir música é fascinante e os horizontes não têm limites.

## OS SONS DO MSX

O MSX pode ser utilizado como um instrumento musical pois possui um circuito interno gerador de sons e, sob a linguagem BASIC, dois comandos que permitem que você o programe para produzir sons sequencialmente, simultaneamente e com velocidade programada. Também pode, com o auxílio de um programa aplicativo e interface, controlar equipamentos que disponham de interface MIDI (Musical Instruments Digital Interface).

Trataremos, neste volume, apenas da primeira aplicação, deixando a segunda para outro volume.

A primeira coisa que devemos aprender neste capítulo é como digitar programas em BASIC que façam seu micro tocar alguma música ou produzir algum som.

Leia cuidadosamente o enunciado de cada exercício a seguir e execute as instruções no seu MSX para ir adquirindo uma certa familiaridade com seu novo "instrumento" musical.

Nesta primeira fase não se preocupe em compreender os programas em si; o exato significado de cada linha do programa em BASIC será entendido com o decorrer do curso.

## EXERCÍCIOS

E 1.1 - Digite o programa da figura 1.2, para ouvir uma demonstração do seu MSX tocando uma antiga melodia em duas vozes. Para facilitar a digitação, comande antes:

KEY 1, "PLAY" + CHR\$(34)

e aperte o RETURN. Desta forma, toda vez que você pressionar a tecla de função F 1, você já terá, na tela, o PLAY com as aspas abertas.

Da mesma forma, comandando:

```
KEY 2,CHR$(34)+CHR$(44)+CHR$(34)
```

você terá, na tecla F2, o "fecha aspas", a vírgula e o "abre aspas" que aparece no meio de cada linha.

Finalmente, comande:

```
KEY 3,CHR$(34)+CHR$(13)
```

Assim fazendo você terá, na tecla F 3, o "fecha aspas" e o RETURN necessários para terminar as linhas que contêm o PLAY. Desative o CAPS LOCK para digitar o que está contido entre aspas em minúsculas. Desta forma evitaremos confundir o O com 0 e o 8 com B.

Após digitar o programa, rode-o com RUN e RETURN, se aparecer uma mensagem de "função ilegal" em X, liste o programa e procure na linha X o erro que você cometeu.

Figura 1.2 - Exemplo de música no MSX.

```
100 INTERVAL ON:SCREEN 2
110 ON INTERVAL=54 GOSUB 480
120 FOR I=0 TO 1
130 PLAY "t200s0m7000","t200s0m7000"
140 PLAY "r4o3","r4o3"
150 PLAY "a4o4c2d4","r4L64ao4cea"
160 PLAY "e4.f8e4","o3ao4cea"
170 PLAY "d2o3b4","o3gbo4dg"
180 PLAY "g4.a8b4","o3gbo4dg"
190 PLAY "o4c2o3a4","o3ao4cea"
200 PLAY "a4.g#8a4","dfao5d"
210 PLAY "b2g#4","o4eg#bo5d"
220 PLAY "e2a4","o4eg#bo5d"
230 PLAY "o4c2d4","o3ao4cea"
240 PLAY "e4.f8e4","o3ao4cea"
250 PLAY "d2o3b4","o3gbo4dg"
260 PLAY "g4.a8b4","o3gbo4dg"
270 PLAY "o4c4.o3b8a4","o3ao4cea"
280 PLAY "g#4.f#8g#4","o4eg#bo5d"
290 PLAY "a2","o3ao4cea"
300 PLAY "a2r4","o3ao4cea"
```

```

310 PLAY "o4g2.", "14o4ceg"
320 PLAY "g4.f8e4", "ceg"
330 PLAY "d2o3b4", "o3gbo4d"
340 PLAY "g4.a8b4", "o3gbo4d"
350 PLAY "o4c2o3a4", "o3ao4ce"
360 PLAY "a4.g#8a4", "o3ao4ce"
370 PLAY "b2g#4", "eg#b"
380 PLAY "e2.", "eg#b"
390 PLAY "o4g2.", "04ceg"
400 PLAY "g4.f8e4", "ceg"
410 PLAY "d2o3b4", "o3gbo4d"
420 PLAY "g4.a8b4", "o3gbo4d"
430 PLAY "o4c4.o3b8a4", "o3ao4ce"
440 PLAY "g#4.f#8g#4", "eg#b"
450 PLAY "a2.a2r4", "o3ao4ceo3a2"
460 NEXT I
470 END
480 X%=256*RND(1) : Y%=192*RND(1)
490 R%= 12*RND(1)+2: C%= 14*RND(1)+2
500 CIRCLE (X%,Y%),R%,C%
510 RETURN

```

Agora responda as seguintes perguntas:

- a) Qual o título da música tocada?
- b) Quem é o autor?
- c) De que século é a música?

As respostas de todos os exercícios estão no apêndice 2.

- E 1.2 - Experimente ligar seu micro num amplificador de 4UD10. Veja no manual de instruções como fazer isso. Toque a música do exercício 1.1 e veja agora a qualidade do som. Para gravar seu programa em fita cassete, após conectar os cabos ao gravador, ligá-lo corretamente e colocar nele uma fita virgem, comande:

CSAVE "FIG1.2"

- E 1.3 - Vamos ver, agora como produzir ruídos no MSX. Para facilitar a digitação, você pode previamente comandar

KEY 1, "SOUND" (e RETURN)

pois a palavra SOUND aparece muitas vezes e é bom associá-la a uma tecla (no caso F1).

Digite o programa da figura 1.3 e rode-o com RUN (e RETURN).

Figura 1.3 - Efeito sonoro no MSX.

```
100 SOUND 7,28
110 SOUND 1,0
120 SOUND 3,0
130 SOUND 6,0
140 SOUND 0,100
150 SOUND 8,13
160 SOUND 9,0
170 SOUND 10,15
180 SOUND 12,255
190 FOR L=1 TO 2000:NEXT L
200 FOR L=100 TO 30 STEP -.04
210 SOUND 0,L
220 NEXT L
230 SOUND 8,0
240 FOR L=2 TO 31 STEP .01
250 SOUND 6,L
260 NEXT L
270 SOUND 10,16
280 SOUND 13,0
```

a) O que lembra o ruído produzido pelo programa da figura 1.3 ?

b) Se você levar o cursor até a linha 190 e alterar o 2000 para 1000, que efeito isso tem sobre a sequência de ruídos? (não esqueça de digitar RETURN após cada alteração de linhas para que ela seja atualizada na memória do micro).

c) Levando o cursor até a linha 240 e alterando o STEP .01 para STEP.1, que alteração isso traz ao programa?

E 1.4- Grave o programa da figura 1.3 em cassete com

CSAVE "FIG1-3"

em drive com:

SAVE "FIG1-3.BAS"







## A NOTAÇÃO MUSICAL PARA A ALTURA

### INTRODUÇÃO

Assim como a linguagem falada dispõe de uma representação escrita, a linguagem musical também dispõe de uma.

Esta representação passou por vários processos de adaptação à linguagem musical de cada época, até atingir a forma atual, conforme veremos neste capítulo.

Antes de atingir a forma atual de representação, vários pesquisadores, ao longo dos séculos, dedicaram-se a essa tarefa.

O sistema utilizado na Grécia Antiga era baseado em símbolos chamados neumas, derivados das letras do alfabeto grego. A notação grega por neumas desenvolveu-se desde o século VII a.C. até o século IV d.C., aproximadamente.

Com o domínio romano, esse sistema foi modificado. Atribui-se a um senador romano do século IV d.C., Alípio, o primeiro tratado romano sobre música. Esse tratado era denominado "As Tábuas", que constava de aproximadamente 1500 sinais para a representação musical.

Até o século X d.C., várias tentativas foram feitas, até surgirem os primeiros conceitos do sistema que utilizamos atualmente.

Mas que tipo de característica de um som pode ser representada graficamente?

É o que veremos a seguir...

## PROPRIEDADES DO SOM

Podemos resumir as características de um som sob quatro aspectos: altura, duração, intensidade e timbre. Estas quatro propriedades representam, de forma bastante simplificada, as características complexas de um sinal sonoro. Algumas características físicas de sinais sonoros serão vistas no volume 2.

Vejamos, então, o que são estas quatro características:

**ALTURA** - corresponde ao posicionamento do som segundo os conceitos de graves, médios e agudos. Digite o programa da figura 2.1.

Figura 2.1 - Sons graves, médios e agudos.

```
10 PLAY"L16V15"  
20 FOR I=1 TO 96  
30 N$="N"+STR$(I)  
40 PLAY N$  
50 NEXT I
```

Você pode perceber sons de diferentes alturas, que vão desde os graves, passando pelos médios e terminados nos agudos. Com este programa, inclusive, além de perceber a diferença de altura, você pode conhecer toda a gama de sons que seu MSX pode tocar, pois o primeiro que você ouve é o mais grave possível e o último é o mais alto (agudo).

Note, também que os sons mais altos, dependendo da sua acuidade auditiva (que entre outras coisas, é também função da idade) podem ser virtualmente inaudíveis!

**DURAÇÃO** - a duração de um som corresponde ao tempo em que o som se prolonga. Para compreender melhor esta característica, digite o programa a seguir.

Figura 2.2 - Duração de sons.

```
10 PLAY"v15"  
20 PLAY"a2r2a16r2"  
30 GOTO 20
```

Podemos dizer que um som é longo e o outro é curto. Este é o conceito de duração de um som. Para breçar a demonstração, basta digitar CONTROL + STOP .

**INTENSIDADE** - é a propriedade de um mesmo som parecer forte ou fraco aos órgãos auditivos. O programa a seguir mostra dois sons que têm intensidades diferentes, apesar de terem a mesma altura e a mesma duração.

Figura 2.3 - Intensidade de sons.

```
10 PLAY"v9a2r2v15a2r2"  
20 GOTO 10
```

**TIMBRE** - é um atributo especial do som. O timbre pode estar associado com o instrumento musical usado ou com as características da fonte sonora. Por exemplo, o soar de um tambor é diferente do de um sino. Um instrumento musical eletrônico pode gerar sons de vários timbres. Digite o programa da figura 2.4.e ouça a mesma musiquinha tocada com dois timbres diferentes.

Figura 2.4 - Timbre de sons.

```
10 PLAY"o3","o4","o8"  
20 B$="L8s0m6000":C$="s13m600"  
30 A$="r1cBdB8f4f8f4c8dBc2d4d8d4cBg8f8e  
4e8e4c8d8e8f4f8f8"  
40 PLAY B$,B$,B$  
50 PLAY A$,A$,A$  
60 PLAY"o4","o4","o6"  
70 PLAY C$,C$,C$  
80 PLAY A$,A$,A$
```

## A NOTAÇÃO MUSICAL

Uma de nossas principais preocupações com o mundo que nos cerca é entender suas mensagens e transmitir nossas informações a outras pessoas. Há vários tipos de linguagens e vários tipos de notações para se transferir as informações.

A notação musical atual tem origens no trabalho de um monge italiano chamado Guido D'Arezzo (995-1050) que adotou um sistema para representar os sons. Nessa época, a música obedecia a determinados fundamentos e os instrumentos musicais já lembravam os atualmente usados em orquestras.

## A NOTAÇÃO PARA A ALTURA

Para representar os sons, Guido D'Arezzo idealizou um conjunto de quatro linhas paralelas, a pauta, onde representava a altura dos sons e dois símbolos: para os instrumentos de sons mais graves (clave de fá) e os instrumentos de sons mais agudos (clave de sol). Os nomes das notas musicais foram tirados das sílabas iniciais de um hino a São João Batista, que os meninos cantores entoavam para protegê-los da rouquidão:

Ut queant laxis resonare fibris

Mira gestorum famuli tuorum

Solve polluti labu reatum

Sancte Ioannes.


A tradução seria:

Para que nós, servos, com nitidez e língua desimpedida, o milagre e a força dos teus feitos elogiemos, tira-nos a grave culpa da língua manchada, São João.

Tínhamos, então, UT, RE, MI, FA, SOL e LA. Por ser difícil de pronunciar, o UT acabou virando DO (de DOMINUS) e mais tarde apareceu o SI (talvez por homenagem a São João - Sancte Ioannes), dando origem à sequência de monossílabos DÓ, RÉ, MI, FÁ, SOL, LÁ e SI, que são os indicadores da altura dos sons, também chamados notas musicais.

A figura 2.5 mostra a forma de representação usada nessa época. As notas eram representadas por quadrados, losangos e traços.

Figura 2.5 - Representação musical de Guido D'Arezzo.



A evolução do trabalho de Guido D'Arezzo representa a notação musical atualmente utilizada para determinarmos a altura dos sons, ou seja, o pentagrama, ou pauta, que é um conjunto de cinco linhas paralelas, igualmente espaçadas, como mostra a fig 2.6.


Figura 2.6 - O pentagrama.



Um dos problemas gerados pelo pentagrama foi a dificuldade para se representar sons muito graves ou muito agudos. Se todos fossem colocados, dos mais graves aos mais agudos, em sequência, teríamos de criar uma pauta com muitas linhas, o que dificultaria a leitura. Foram desenvolvidos alguns sinais, chaves (alguns chamam de chaves), para indicar se os sons representados no pentagrama eram mais graves ou mais agudos. Esses sinais são sempre colocados no início de cada pentagrama, mas podem ser, às vezes, trocados durante a sequência das notas no pentagrama.

Para os instrumentos de sons graves, utilizamos a Clave de Fá e para os instrumentos de som agudo, a Clave de Sol, como mostra a figura 2.7. Para assinalar a altura de cada som, utilizamos um símbolo ovalado (o).

Figura 2.7 - As chaves de Fá e Sol.



NOTA: Existe outra chave, a de Dó, mas é pouco utilizada e não trataremos dela, nem da Clave de Sol na primeira linha, nem da Clave de Fá na terceira linha.

A Clave de Sol dá à nota da segunda linha o nome de Sol. A clave de Fá dá à nota da quarta linha o nome de Fá. A relação entre as duas claves é tal que as duas representações da figura 2.8 mostram a nota Dó com a mesma altura na Clave de Fá e na Clave de Sol.


Figura 2.8 - Correlação entre as claves.




Vejamos como é a sequência de notas nas duas claves:

Figura 2.9 - Sequências nas Claves de Sol e Fá.

CLAVE DE SOL




CLAVE DE FÁ



Bem, você deve ter percebido que quando sons de determinada altura não "cabem" dentro do pentagrama, determinamos outras linhas, acima ou abaixo. Essas linhas são chamadas linhas suplementares, como mostra a figura 2.10

Figura 2.10 - Linhas suplementares.



O programa da figura 2.11 vai lhe mostrar como são as alturas correspondentes às representações nas Claves de Fá e Sol. Através das setas para cima e para baixo você poderá mudar a altura da nota e percorrer a extensão de dois pentagramas. Apertando a barra de espaços você pode ouvir o som correspondente.

Figura 2.11 - Alturas nas Claves de Sol e Fá.

```

100 P$(1)="DO      C":P$(2)="RE      D"
110 P$(3)="MI      E":P$(4)="FA      F"
120 P$(5)="SOL     G":P$(6)="LA      A"
130 P$(7)="SI      B"
140 COLOR 15,1,1:SCREEN 2,,0
150 OPEN"GRP:"AS #1
160 PRESET(31,170)
170 PRINT#1,"OITAVA  NOTA  CIFRA"
180 FOR C=0 TO 1
190 FOR L=63+48*C TO 95+48*C STEP 8
200 LINE(0,L)-(255,L)
210 NEXT L,C
220 FOR I=1 TO 26
230 D=23+8*I
240 PRESET(D,152-4*I):PRINT#1,CHR$(1)+
    CHR$(73)
250 LINE(D,5)-(D+6,21),,B
260 IF IMOD7=2 OR IMOD7=6 THEN 280
270 LINE(D-3,5)-(D+1,15),,BF
280 NEXT I
290 LINE(30,151)-(38,151):LINE(126,103)
    -(134,103):LINE(222,55)-(237,55)
300 PRESET(14,86):DRAW"C15S12LHUERM+2,+1
    M+1,+2M-1,+2M-2,+1M-2,-1H2U2M+1,-2E4U2HD
    16GH"
310 PRESET(14,87):DRAW"C15S12LHUERM+2,+1
    M+1,+2M-1,+2M-2,+1M-2,-1H2U2M+1,-2E4U2HD
    16GH"
320 PRESET(14,120):DRAW"C15S12LHUERM+2,+
    1M+1,+2M-1,+3G2":PSET(27,115):PSET(27,12
    3)
330 PRESET(14,121):DRAW"C15S12LHUERM+2,+
    1M+1,+2M-1,+3G2":PSET(27,115):PSET(27,12
    3)
340 FOR T=1 TO 8:C%=C%+CHR$(PEEK(7222+T)
    ):D%=D%+CHR$(PEEK(7206+T)):NEXT T:SPRITE
    $(1)=C%:SPRITE$(2)=D%
350 I=13:PLAY"S0M7000"
360 A=STICK(0)
370 I=I+(A=5)-(A=1):X=23+8*I:Y=152-4*I

```

```

380 IF I<1 THEN I=1
390 IF I>26 THEN I=26
400 O=(I+15)\7:W=(I+15)MOD7+1
410 IF A=0 THEN 440
420 LINE (40,180)-(240,190),1,BF
430 PRESET(41,181):PRINT#1,0;"      ";P$(
W)
440 PUT SPRITE 3,(X,Y-1),,1
450 PUT SPRITE 4,(X,22),,2
460 N=12*(I\7)+14+2*(IMOD7)+(IMOD7>1)+(I
MOD7>5)
470 A$="N"+STR$(N)
480 IF STRIG(0) THEN GOSUB 500
490 GOTO 360
500 PLAY A$
510 IF STRIG(0) THEN 510
520 RETURN

```

## A EXTENSÃO DOS INSTRUMENTOS MUSICAIS

Cada instrumento musical, além da característica particular de timbre, possui também uma característica chamada extensão, que é o intervalo entre as alturas mínima e máxima que tal instrumento alcança.

Por exemplo, você pode observar a seguir as extensões utilizadas em concertos, para alguns instrumentos de sopro.

Figura 2.12 - Extensão de alguns instrumentos.






O arranjo e orquestração de uma peça musical dependem da intensidade, extensão e timbre dos instrumentos.

Como veremos no capítulo 6 o menor intervalo que separa duas notas do mesmo nome, numa sucessão de Dó a Dó, Ré a Ré, etc... é chamada de oitava. Chamamos às sucessões de notas de escalas.

A escala geral representa, em termos de altura, a totalidade dos sons e sua extensão é de oito oitavas, designadas por Dó-0 a Dó-9. O MSX permite gerar sons desde o Dó-1 até o Si-8, conforme mostra a figura 2.13.

Figura 2.13- Extensão do MSX



Para você poder comparar, a extensão de um piano vai de Lá-0 a Dó-8. O programa da figura 2.14 pode demonstrar a extensão do MSX.

Figura 2.14 - Extensão do MSX.

```
10 PLAY"=0m8000"
20 FOR I=1 TO 8
30 O$="o"+STR$(I)
40 PLAY O$
50 PLAY "cdefgab"
60 NEXT I
```

## A EXTENSÃO DA VOZ HUMANA

A voz humana, que é gerada por um processo bastante complexo envolvendo a cavidade bucal, cordas vocais, traquéia e muitos outros órgãos, pode atingir extensões incríveis, como por exemplo a da peruana Yuma Sumac.

Geralmente os homens têm altura de voz mais grave que as mulheres e crianças. A voz mais aguda é chamada soprano e a mais grave, baixo. São classificadas, da mais grave à mais aguda em: baixo, barítono, tenor, contralto, meio soprano e soprano.

Vejamos como é a representação da extensão para cada uma destas vozes na figura 2.15.

Figura 2.15- Extensão da voz humana.



Assim, as vozes graves barítono e baixo são normalmente escritas na clave de fá e as vozes tenor, contralto, meio soprano e soprano, na clave de sol.

Vamos descobrir a extensão de sua voz? Não é difícil. O programa da figura 2.11 pode ajudá-lo a descobrir. Basta cantar junto com o MSX. Comece com a nota mais grave que você pode cantar. Compare com o som gerado no MSX. Tente a mais aguda e compare. Pronto. Basta olhar a figura 2.15 e você descobrirá!

## AS NOTAÇÕES ALEMÃ E INGLESA

Devido à origem dos nomes das notas musicais, muitos estudiosos e pesquisadores da música de origem não latina continuaram a utilizar uma nomenclatura muito comum no continente europeu. As notas até o século XII DC eram também conhecidas pelas letras A, B,

C, D, E, F e G. A correspondência entre os nomes em latim e as letras foi elaborada de modo que a letra A corresponde a Lá, B a Si e, em sequência, até G, que corresponde a Sol. Embora a nomenclatura definida por Guido D'Arezzo seja muito utilizada, nos países de língua anglo-saxônica a representação por letras perdura até hoje. Essa é a representação utilizada pelo comando PLAY, no MSX.

## EXERCÍCIOS

E 2.1 - Digite o programa da figura 2.16. Ao rodá-lo você ouvirá dois sons diferentes.

Figura 2.16

```
10 PLAY"v15L4o4"  
20 PLAY"crbr"  
30 GOTO 20
```

O que distingue os dois sons é:

- a) Intensidade
- b) Altura
- c) Duração
- d) Timbre

E 2.2 - Digite o programa da figura 2.17. Ao rodá-lo você ouvirá vários sons diferentes.

Figura 2.17

```
10 FOR X=0 TO 1000 STEP .314  
20 V%=STR$(INT((COS(X)*4)+11))  
30 PLAY "o4v"+V%+"b4r4"  
40 NEXT X
```

O que os distingue é:

- a) Intensidade
- b) Altura
- c) Duração
- d) Timbre

E 2.3 - Digite o programa da figura 2.18 e rode-o. Ele

simula a emissão de uma mensagem em código MORSE (se você for radio-amador não tente entender pois a emissão é aleatória).

Figura 2.18

```
10 PLAY"04V15T200"  
20 N=INT(RND(-TIME)*4)+1  
30 FOR I=1 TO N  
40 X=8^(INT(RND(-TIME)+.5)+1)  
50 L$="L"+STR$(X/2)  
60 PLAY L$+"BR64"  
70 NEXT I  
80 PLAY "R4"  
90 GOTO 20
```

Qual a qualidade do som que permite, a um rádio-telegrafista, distinguir entre o "ponto" e a "linha" ?

- a) Timbre
- b) Intensidade
- c) Duração
- d) Altura

E 2.4 - O comando PLAY permite fazer com que o MSX toque uma sequência de notas de diferentes alturas, usando a nomenclatura Alemã;

Dó = C  
Ré = D  
Mi = E  
Fá = F  
Sol = G  
Lá = A  
Si = B


Vamos agora aprender a transcrição de uma "partitura". Por enquanto não iremos nos preocupar com a duração das notas mas apenas com sua altura.

Digite a linha

```
10 PLAY " 04 S0 M 5000 L 4 "
```

que configura a primeira voz de seu MSX para timbre de "piano" (não se preocupe em entendê-la por enquanto).  
Veja agora a partitura da figura 2.19.

Figura 2.19



Como você pode ver com a ajuda na página 46, a sequência de notas indicada é

Sol.Fá.Mi.Ré.Mi.Fá.Sol.Sol.Sol  
G. F. E. D. E. F. G. G. G

Para fazer seu MSX tocar esta sequência, basta digitar:

20 PLAY "GFEDEFGGG"

Aparte o "ritmo" que não está muito de acordo, você pode reconhecer o começo de uma cantiga infantil. Qual é?

E 2.5 - Vamos continuar a cantiga! Veja a figura 2.20 e digite a linha 30 a ela correspondente.

Figura 2.20



30 PLAY " .....

E 2.6 - Continuando, veja a figura 2.21 e digite a linha 40.

Figura 2.21




40 PLAY " .....

E 2.7 - Finalizando, veja a figura 2.22 e complete seu programa com a linha 50.

Figura 2.22



50 PLAY " .....





## A NOTAÇÃO MUSICAL PARA A DURAÇÃO

### AS FIGURAS

A notação musical para duração dos sons tem origens no Mensuralismo, teoria desenvolvida por Walter Oddington e Franco de Colônia no século XII. A idéia era criar símbolos diferentes para uma mesma indicação de altura, de modo a termos durações bem precisas.

A evolução deste sistema representa o sistema atualmente utilizado. A notação musical é composta, então, por sinais colocados no pentagrama, que identificam, segundo seu posicionamento no pentagrama, sons mais graves e mais agudos (comumente chamados notas) e, segundo a sua forma, sons de duração mais curta e mais longa (chamados de figuras).

Os símbolos para a duração são:

Figura 3.1 - Símbolos para a duração.



Dependendo do posicionamento no pentagrama, estes símbolos podem aparecer também na forma,

Figura 3.2 - Outra forma dos símbolos de duração.



Há uma correlação matemática entre a duração relativa das figuras, como veremos a seguir.

Figura 3.3 - Relação entre as durações.




Assim, a duração de uma semibreve é igual à duração de duas mínimas, igual à duração de quatro semínimas e assim por diante. No MSX existem 2 maneiras de se definir a duração de um som: na "string" contida entre as aspas que vêm depois de um comando PLAY podemos determinar a duração de um grupo de sons colocando à sua frente o sub-comando,

Lx

onde x é um número entre 1 e 64 que determina a duração das figuras especificadas após este comando.

A correspondência entre as durações e o valor de x é dada na figura 3.4.

Figura 3.4 - Relação entre as durações e o valor de L.





Digite o programa da figura 3.5 para perceber este efeito.

Figura 3.5

```
10 PLAY"L1cdefgabo5co4bagfedc"  
20 PLAY"L4cdefgabo5co4bagfedc"  
30 PLAY"L16cdefgabo5co4bagfedc"  
40 PLAY"L64cdefgabo5co4bagfedc"
```

Note que as linhas são praticamente iguais: basta digitar a primeira e depois levar o cursor para alterar o número da linha (e o valor de x). Não esqueça de digitar RETURN após cada alteração. Dê um LIST para se certificar que as 4 linhas estão na memória do micro.

A outra maneira é a de se colocar o x logo após a letra (cifra) que indica a nota. Se, após a letra, o x não for especificado, vale o último Lx utilizado.

Digite o programa da figura 3.6 que representa uma melhoria do que você fez nos exercícios de E2.4 a E2.7 da aula passada.

Figura 3.6

```
10 PLAY "o4s0m5000LB"  
20 PLAY "GFEDEFG4G4G4"  
30 PLAY"AGF4F4F4GFE4E4E4"  
40 PLAY"CCA4A4A4BAG4G4G4"  
50 PLAY"EFG4EFGFEDC2o6C"
```

Note que no final da linha 10 foi especificado LB que faz todas as notas sem valor de X serem colcheias. Nas outras linhas, algumas figuras foram alteradas para semínimas e a última (antes do "apito") virou mínima!

Como a relação das durações é de múltiplos de dois, utilizamos um ponto à frente da figura, chamado ponto de aumento, quando precisamos de uma relação de múltiplos de três. Este ponto aumenta a duração do som de metade do seu valor. Suponha que, num determinado contexto, tenhamos:

$o \rightarrow 2$  segundos, então  $o. \rightarrow 2+2/2 = 3$  segundos.


Podemos colocar um segundo ponto, se quisermos mais

aumento. O aumento será de metade da duração do primeiro ponto. Assim, se:

$o \rightarrow 2 \text{ seg.}$ , então  $o.. \rightarrow 2+2/2+2/2*1/2 = 3*1/2 \text{ seg.}$

Há uma outra forma de se aumentar a duração de um som. Utilizamos os símbolos  $\frown$  e  $\smile$ , chamados ligaduras, para unir a duração de duas figuras. Portanto,

Figura 3.7



A utilização da ligadura ou do ponto depende do contexto musical, como veremos no decorrer deste capítulo.

No MSX também podemos utilizar o ponto. Digite o programa da figura 3.8 para perceber isso.

Figura 3.8


```
10 SCREEN 1:INTERVAL ON
20 PLAY "v15"
30 X=120:KEYOFF
40 PRINT TAB(2);"NOTA";
50 PRINT TAB(10);"DURACAO"
60 FOR I=0 TO 4
70 A$="a1"+STRING$(I,".")
80 X=X+120/(2^I)+(I=0)*120
90 LOCATE 2,4*I+2
100 PRINT A$:TIME=0:PLAY A$
110 LOCATE 10,4*I+2:T=TIME/60
120 PRINT USING"##.##";T
130 IF TIME<X THEN GOTO 110
140 FOR F=0 TO 1000:NEXT F
150 NEXT I
```

No programa a duração é dada até a casa dos décimos de segundo, indicando um valor aproximado. Apesar disso você pode notar que cada ponto implica um acréscimo na duração igual à metade do acréscimo anterior.

## PAUSAS

Suponha que, no meio de uma execução, seja necessário um intervalo de silêncio entre um som e outro. Para representar esses intervalos de silêncio foram criadas as pausas, correspondentes ao tempo de duração dos sons. Os símbolos das pausas são vistos na figura 3.9.

Figura 3.9 - Símbolos das pausas



No MSX a pausa é especificada pelo subcomando Rx, onde x segue à convenção da duração das figuras.

## ALGUMAS CONVENÇÕES

A partir destes conceitos, poderemos representar a altura e a duração dos sons no pentagrama, como no exemplo da figura 3.10.

Figura 3.10



É costume, para facilitar a escrita e a leitura, escrever-se, da terceira linha para baixo, as representações com a haste voltada para cima e, da terceira linha para cima, a haste voltada para baixo.

Em certas representações, para facilitar o agrupamento de colcheias, semicolcheias, fusas e semifusas, é comum vê-las unidas. As relações são exemplificadas na figura 3.11.

Figura 3.11



## AS NOTAÇÕES PARA INTENSIDADE E TIMBRE

As outras propriedades do som, que faltam ser identificadas no pentagrama são intensidade e timbre.

A representação de intensidade está ligada a um aspecto da música chamado dinâmica. A dinâmica é a variação da intensidade dos sons ao longo de uma música. A notação utilizada não é uma medida quantitativa, ou seja, não é representada através de números, mas de palavras oriundas do italiano. É um costume, portanto vamos descrevê-los.

SÍMBOLO	NOME	SIGNIFICADO
pp	pianissimo	muito leve
p	piano	leve
mf	mezzo forte	meio forte
f	forte	forte
ff	fortissimo	muito forte

Há outras muitas indicações, como crescendo, decrescendo, os sinais < (aumentando) e > (diminuindo) que podem aparecer ao longo de uma partitura. Com um pouco de bom senso e algumas associações, não é difícil "decifrá-los".

Para os diferentes timbres, dependendo do tipo de música composta ou executada, há várias formas de representá-los, a maioria com palavras ou símbolos. Por exemplo, a figura 3.12 mostra um trecho de uma partitura da compositora Wendy Carlos.

Os timbres são os dos instrumentos escritos à esquerda dos pentagramas.

Figura 3.12



## AS REPRESENTAÇÕES NO MSX

Como já vimos, a duração relativa de uma nota, no MSX é dada, no comando PLAY pelo subcomando Lx.

A duração absoluta é definida pelo subcomando Tn. Aumentando n, a duração absoluta dos sons diminui. O valor n vai de 32 a 255.

A intensidade de uma nota só pode ser alterada pelo valor Vn, onde n pode assumir um valor qualquer de 0 a 15.

Digite o programa da figura 3.13 e veja como podemos mudar a velocidade de execução e o volume ao longo da música.

Se você gravou o programa da figura 3.6, carregue-o e faça as alterações necessárias.


Figura 3.13

```
5 N=0
10 PLAY "o4L8T100"
20 PLAY "o4V8G4FEDEFG4G4G4"
30 PLAY"V10AGF4F4F4GFE4E4E4"
40 PLAY"V12CCA4A4A4BAG4G4G4"
50 PLAY"V15EFG4EFGFEDC2o6C"
60 IF N=1 THEN END
70 N=1:PLAY"r1"
80 PLAY "T240"
90 GOTO 20
```

Há uma forma de alterarmos levemente os timbres do comando PLAY, através dos subcomandos Mn e Sn. O subcomando Mn define o período da variação do volume durante a execução da nota e o subcomando Sn o formato do envelope.

Mas o que é envelope? Esse conceito vem de uma característica muito interessante dos instrumentos musicais. Se você tocar uma nota qualquer ao piano, perceberá que a intensidade no instante em que você aperta a tecla é grande e vai decrescendo com o decorrer do tempo até desaparecer completamente: mesmo que você a mantenha pressionada. O gráfico da figura 3.14 mostra, de forma simplificada, o envelope do som gerado quando você toca uma tecla de piano.


Figura 3.14



Todos os instrumentos de percussão, como o tambor, pratos, triângulos, e os instrumentos em que a corda é puxada ou percutida, como o violão, o cravo e o piano têm um envelope parecido.

Os instrumentos de sopro dependem do esforço gerado pelo executante, como a flauta, o trombone, etc... O violino tem um envelope que depende do movimento do arco sobre a corda. O órgão de tubos, como tem um som que é praticamente contínuo, pois trabalha com compressores de ar, tem um envelope como o da figura 3.15. O som cessa quando você solta a tecla.

Figura 3.15




Em geral, os envelopes são simples mas, nos instrumentos eletrônicos, podemos ajustá-los, conforme alguns parâmetros mais conhecidos pela sua representação em inglês, que são:

- delay (atraso inicial)
- attack (ataque)
- decay (decaimento)
- sustain (sustentação)
- release (amortecimento)

Alguns instrumentos musicais eletrônicos dispõem de um outro conceito, ao invés desses quatro parâmetros, podemos definir apenas dois: nível (N) e inclinação (uma maior inclinação define se a intensidade aumenta ou diminui mais rapidamente e o nível define os graus de intensidade em cada ponto. Podemos, dependendo do número de níveis, definir envelopes como o da figura 3.16.

Figura 3.16



Dependendo do instrumento, o envelope pode, além de controlar a intensidade (ou volume), controlar outros valores como: brilho, afinação relativa, timbre, etc...


Bem, se isso tudo parecer muito complicado, não se preocupe, mais algumas páginas e você já poderá transcrever algumas músicas para o MSX.

## EXERCÍCIOS

E 3.1- Quando você transcreve uma partitura musical no computador usando a notação por cifras (CDEFGA B) há necessidade de especificar a que oitava

essas notas pertencem. Para isso basta usar o comando `ox` onde `x` é o número da oitava, que pode variar de 1 a 8. A grande maioria das músicas que você transcreverá, porém, se situa entre a oitava 2 e a 5. Ao ligarmos o MSX ele utiliza como valor padrão `o4` (valor "default"). Uma vez especificada a oitava, todas as notas transcritas serão consideradas dentro desta oitava até que ocorra nova especificação. A figura 3.17 serve como orientação para verificar a que oitava pertence cada grupo de 7 notas (CDEFGAB). A nota C assinalada no centro da figura (C a oitava 4) é denominada **Dó CENTRAL**.

Figura 3.17




Na figura 3.18 estão assinaladas 6 notas. Para transcrevê-las no MSX precisamos especificar sua oitava e sua cifra.



A nota nº 1, por exemplo, será tocada pelo comando

Nota 1: PLAY" n2E"

Figura 3.18




Complete a relação a seguir:

nota 2: PLAY"...  
nota 3: PLAY"...  
nota 4: PLAY"...  
nota 5: PLAY"...  
nota 6: PLAY"..."

E 3.2- Vamos, agora, aprender a transcrever músicas um pouco mais complexas. Vamos começar por um "moteto" do século XIII. A figura 3.19 fornece a partitura e você deverá gerar um programa no MSX que a toque. Existe apenas uma particularidade, nesta partitura, que será comentada numa aula posterior: quando, antes da figura que representa a nota aparecer o símbolo  $\flat$ , devemos digitar um sinal de "menos" (-), logo após a cifra (e antes do número que determina a duração). Na partitura da figura 3.19, onde ocorre este sinal você deverá digitar B-4. A cada divisão vertical, comece um novo comando PLAY.

Figura 3.19



Com base nesta partitura e usando a figura 3.17 como referência, complete o programa da figura 3.20. A linha 5 serve para produzir um envelope de piano (nos próximos exercícios veremos isso com mais detalhes).

Figura 3.20

```

5  PLAY "s0m8000","s0m8000","s0m8000"
10 PLAY "o5d8d8c4o4a4"
20 PLAY "o5c8c8o4b4o5c4"
30 PLAY "d2r4"
40 PLAY "...
50 PLAY "...
60 PLAY "...
70 PLAY "...
80 PLAY "...
90 PLAY "...

```

E 3.3- O seu MSX, na realidade, possui 3 vozes para tocar músicas. A sintaxe completa do comando PLAY é:

PLAY"primeira voz","segunda voz","terceira voz"

Vamos, então, complementar o programa que você digitou no exercício E 3.2, inserindo uma segunda voz baseada na partitura da figura 3.21. Para sua orientação foi colocado o número da linha acima de cada divisão.

Figura 3.21



Baseando-se no programa digitado no exercício anterior, complete-o como indicado na figura 3.22 .

Figura 3.22


```

5  PLAY "s0m8000","s0m8000"
10 PLAY "o5d8d8c4o4a4","o4d4a2"
20 PLAY "o5c8c8o4b4o5c4","a4a2"
30 PLAY "d2r4","..."
40 PLAY "c8o4b8a4g4","..."
50 PLAY "a8g8a4b4","..."
60 PLAY "o5c2r4","..."
70 PLAY "d8d8o4b-4g4","..."
80 PLAY "o5d8d8c4o4a4","..."
90 PLAY "a2r4","..."

```

E 3.4- Vamos, agora, introduzir uma 3ª voz em nosso "moteto" do século XIII. A partitura da terceira voz é dada na figura 3.23. Cuidado que ela está em clave de Fá!

Figura 3.23



Com base no seu programa em 2 vozes do exercício E 3.3, complete a terceira voz na figura 3.24. Para não ter problemas na inserção de mais comandos numa mesma linha, comande WIDTH 40 ou use a tecla INSERT para colocar o cursor no modo "inserção".

Figura 3.24


```

5  PLAY "s0m8000","s0m8000","s0m8000"
10 PLAY "o5d8d8c4o4a4","o4d4a2","o3d2-"
20 PLAY "o5c8c8o4b4o5c4","a4a2","f2-"
30 PLAY "d2r4","a4a2","..."
40 PLAY "c8o4b8a4g4","g2r4","..."
50 PLAY "a8g8a4b4","a4g2","..."
60 PLAY "o5c2r4","a4g2","..."
70 PLAY "d8d8o4b-4g4","a4b-2","..."
80 PLAY "o5d8d8c4o4a4","a4a4g4","..."
90 PLAY "a2r4","f8e8d2","...."

```

- E 3.5- Pague a listagem do programa apresentada na figura 3.6 e transcreva-a para a partitura da figura 3.25. Lembre-se que, como foi especificado L8 no começo do programa, todas as cifras que não forem seguidas de especificação de duração, são colcheias. Não precisa colocar o apitinho final (o6 C).

Figura 3.25



- E 3.6- No MSX a "forma" do envelope é dada pelo subcomando Sn, onde n é um número entre 0 e 15. O período ou seja a escala de tempo ao longo da qual o envelope tem efeito, é dada pelo subcomando Mx, onde x é um número entre 0 e 65535.

A forma dos envelopes é dada pela figura 3.26. Digite o programa da figura 3.27 e estude o efeito que a variação dos atributos de s e m produz sobre a nota tocada.

Rode o programa várias vezes para estudar com detalhes, mantendo sempre sua atenção na figura 3.26, para associar cada forma (e período) de envelope com o som correspondente. Se quiser estudar um S e um M em particular, quando ele aparecer breque o programa com CONTROL + STOP e leve (usando as setas) o cursor até a linha em estudo. Pressione RETURN. O que acontece?

Figura 3.26 - As formas dos envelopes









VALOR ATRIBUÍDO AO REGISTRO 13 (valores atribuídos a Sn da função PLAY)	FORMA DO ENVELOPE
0,1,2,3 ou 9	
4,5,6,7 ou 15	
8	
10	
11	
12	
13	
14	

Figura 3.27 - Programa de demonstração dos envelopes

```

10 SCREEN 1:KEY OFF
20 FOR S=0 TO 15
30 FOR E=6 TO 16
40 A$="s"+STR$(S)+" m"+STR$(2^E-1)+" c1"
50 PLAY A$:PRINT "PLAY"+CHR$(34)+A$:PRIN
T
60 FOR T=0 TO 2000:NEXT T
70 NEXT E
80 PRINT
90 NEXT S

```



EFGAB CDEFGAB CDEFGAB CDEFGAB



## A DIVISÃO DA MÚSICA

Como já vimos, as figuras definem a duração dos sons, bem como as pausas, o silêncio. Com as dificuldades advindas na representação para vários instrumentos, e para músicas longas, um novo conceito foi introduzido, para facilitar o estudo e a escrita das músicas: o compasso. A idéia é dividir a música em pequenas células, de modo que a duração das células seja igual ou bem definida.

### TEMPOS E COMPASSOS

Cada uma dessas células, o compasso, é formada de um número inteiro de subdivisões, ou unidades chamadas tempos. Temos, mais comumente, os compassos de 2, 3, 4, 5 e 7 tempos, chamados compassos simples.

A existência desse número de tempos se prende a características rítmicas das músicas, que serão oportunamente discutidas. Assumiremos que ritmicamente esses tempos podem soar fortes ou fracos.

Há outras formas, chamadas compostas, cujo número de tempos é obtido multiplicando-se por três o dos compassos simples.

As formas tradicionais de compassos simples obedecem a uma certa lógica, como você poderá observar, digitando o programa da figura 4.1. Quando pedir QUAL COMPASSO (2/3/4/5/7)?

Figura 4.1

```

100 CLEAR 500
110 KEY(1) ON
120 SCREEN 0,,0:KEY OFF
130 ON KEY GOSUB 370
140 C0$="PRESSIONE F1 PARA MUDAR"
150 T0$="V15R5C32"
160 C1$=CHR$(219)+" ":T1$="V15R8..C32"
170 C2$=CHR$(220)+" ":T2$="V11R8..C32"
180 C3$=CHR$(192)+" ":T3$="V7R8..C32"
190 L$(2)=T0$+T3$
200 L$(3)=T0$+T3$+T3$
210 L$(4)=T0$+T3$+T2$+T3$
220 L$(5)=T0$+T3$+T3$+T1$+T3$
230 L$(7)=T0$+T3$+T2$+T3$+T1$+T3$+T3$
240 P$(2)=C1$+C3$
250 P$(3)=C1$+C3$+C3$
260 P$(4)=C1$+C3$+C2$+C3$
270 P$(5)=C1$+C3$+C3$+C1$+C3$
280 P$(7)=C1$+C3$+C2$+C3$+C1$+C3$+C3$
290 LOCATE 3,10
300 INPUT"QUAL COMPASSO(2/3/4/5/7)";A$
310 A=VAL(A$)
320 IF A<2 OR A>7 OR A=6 THEN 290
330 LOCATE 2,15:PRINT A;" ";P$(A)
340 LOCATE 3,17:PRINT C0$:PRINT
350 PLAY L$(A)
360 GOTO 350
370 LOCATE 3,19
380 PRINT"AGUARDE ESVAZIAR O BUFFER"
390 IF PLAY(1)=-1 THEN GOTO 390
400 CLS
410 RETURN 290

```

digite um dos números sugeridos (e RETURN).

- 2 - 1 tempo forte e 1 tempo fraco.
- 3 - 1 tempo forte e 2 tempos fracos.
- 4 - 1 tempo forte, 1 fraco, 1 meio-forte e 1 fraco.
- 5 - 1 tempo forte, 2 fracos, 1 forte e 1 fraco.
- 7 - 1 tempo forte, 1 fraco, 1 meio-forte e 1 fraco,  
1 forte e 2 fracos.

NOTA: Os compassos de 5 e 7 tempos são formados por outros compassos:

- 5 tempos - 1 ternário + 1 binário (ou vice-versa).
- 7 tempos - 1 quaternário + 1 ternário (ou vice-versa).



Digitando F1, após uma certa demora (para esvaziar o buffer do Play), você pode escolher outra opção.

Para parar o programa, digite CONTROL + STOP. Os compassos compostos são formados multiplicando-se por três o número de tempos dos compassos simples.

É costume chamar os compassos, segundo seu número de tempos, conforme mostra a figura 4.2.

FIGURA 4.2 - Nomenclaturas dos compassos.

#### SIMPLES

nº de tempos	nome
2	binário
3	ternário
4	quaternário
5	quinário
7	setenário

#### COMPOSTOS

nº de tempos	nome
6	binário composto
9	ternário composto
12	quaternário composto
15	quinário composto
21	setenário composto

Após tantas definições resta saber como utilizá-las. A idéia é, através de traços verticais, dividir o pentagrama em células de igual duração, como vemos na figura 4.3.

FIGURA 4.3 - Divisão do pentagrama em compassos.



No início do pentagrama, vemos a clave e no fim, um símbolo indicador de fim de trecho.

Colocaremos as figuras em cada compasso, de modo que em cada um deles a duração dessas figuras seja a mesma.

A representação musical para a indicação do tipo de compasso é feita da seguinte forma:








número 1/número 2

número 1 - indica quantos tempos há no compasso.

número 2 - indica qual figura corresponde a um tempo do compasso.

A correspondência, entre número 2 e figura advém da relação matemática entre as figuras como já vimos no subcomando Lx do PLAY, conforme observamos na figura 4.4.

FIGURA 4.4 - Relações das figuras.

FIGURA	NOME	NÚMERO ASSOCIADO
	SEMIBREVE	1
	MÍNIMA	2
	SEMÍNIMA	4
	COLCHEIA	8
	SEMICOLCHEIA	16
	FUSA	32
	SEMIFUSA	64

Portanto, se considerarmos a semibreve como unidade, ou seja, com o valor 1, obteremos a tabela da figura 4.4. A duração de duas mínimas corresponde à de uma semibreve, bem como a de quatro semínimas, oito colcheias, etc...

O valor número 2 é chamado unidade de tempo.


Para representarmos um compasso, escolhemos a unidade de tempo e o tipo do compasso.

Exemplo 1: compasso quaternário, unidade de tempo, semínima, ou seja, quatro tempos de semínima.  
Representação: 4/4

Exemplo 2, compasso ternário, unidade de tempo: colcheia, ou seja, três tempos de colcheia. Representação: 3/8

OBSERVAÇÃO: Muitas vezes o compasso 4/4 é também representado pelo símbolo indicado na figura 4.5.

Figura 4.5



A figura 4.6, por exemplo, mostra os primeiros 3 compassos d' "O GUARANI" de Carlos Gomes.

Figura 4.6




Como o compasso é 4/4 devem caber, em cada célula 4 semínimas, o que é fácil verificar no terceiro compasso.

Vamos subdividir o compasso em 4 tempos (cada um com a duração de uma semínima) e cada tempo em 4 sub-divisões (equivalendo, portanto cada uma a uma semicolcheia).

A representação do compasso 1, neste "diagrama de tempos" é representada na figura 4.7.

Figura 4.7



Como você pode observar, a primeira figura é uma mínima, consequentemente ocupa 2 tempos do nosso compasso. A segunda figura é uma semínima (1 tempo) com um ponto (+ 1/2 tempo). A terceira e a quarta figuras são semicolcheias e cada uma, portanto, ocupará 1/4 de tempo.

O programa, em BASIC, que toca os 3 compassos da figura 4.6, está listado na figura 4.8.

Figura 4.8 - Alguns compassos d'"O GUARANI".

```

5  PLAY "s0m5000"
10 PLAY "o5e2a4.g16f16"
20 PLAY "o5e2a4.g16f16"
30 PLAY "e4a4a4o6c4"

```

Na figura 4.9 está representada a pauta musical dos próximos 4 compassos d'"O GUARANI".

Figura 4.9 - Mais alguns compassos d'"O GUARANI".



Se quisermos transcrevê-los no MSX, nos deparamos com uma novidade: já no compasso 4 existem dois grupos de 3 figuras (assinalados com um algarismo "3" logo abaixo) que não são, a rigor, colcheias como está representado.

Se fossem, o nosso compasso "estouraria" (faça as contas!).

Já sabemos que multiplicando por 2 o atributo Lx, a duração cai pela metade (L8 tem a metade da duração L4). Consequentemente, multiplicando por 3, a duração fica reduzida a 1/3 !

Portanto, aquelas "colcheias estranhas" têm duração L12. A transcrição dos 3 compassos adicionais d'"O GUARANI", ficará como mostrado na figura 4.10.

Figura 4.10

```

5  PLAY "s0m5000"
10 PLAY "o5e2a4.g16f16"
20 PLAY "o5e2a4.g16f16"
30 PLAY "e4a4a4o6c4"
40 PLAY "o5b12o6c12o5b12a12g12a12b4g4"
50 PLAY "o5e2a4.g16f16"
60 PLAY "e2o4a4.g16f16"

```

## ANDAMENTO

Após conhecermos esta forma de dividir a música, resta saber como indicar quando uma música é lenta ou rápida.

Podemos, então, indicá-la através de palavras como: lenta, muito lenta, rápida, moderada, etc. Estes termos são utilizados, só que em italiano, para preservar as abreviações das notações clássicas.

Com a evolução da mecânica de relojoaria, em 1816, Maelzel desenvolveu um dispositivo mecânico que marcava, com precisão, os tempos com um sinal sonoro, o metrônomo. Atualmente dispomos de metrônimos eletrônicos.

O metrônomo é graduado com divisões de 40 a 208, ou seja, emite de 40 a 208 sinais sonoros por minuto. Podemos, assim, associar uma determinada figura a esses sinais sonoros e assim determinar o andamento ou velocidade da música. As divisões do metrônomo não são contínuas, de um em um, mas segundo intervalos, por convenção.

Assim, nas músicas, representamos a velocidade, ou andamento com notações do tipo da figura 4.11.

Figura 4.11

$$\left( \text{semibreve} \right) = 120$$

ou seja, regulamos o metrônomo em 120 batidas por minuto e cada batida corresponde à execução de uma semínima.

Há uma correlação entre nomenclatura em italiano e as marcações do metrônomo, como podemos ver:

**Andamentos Vagarosos:**

Grave, Largo, Lento, Adagio e Larghetto vão de 40 a 72 tempos por minuto.

**Andamentos Moderados:**

Moderato, Andante, Andantino e Allegretto de 72 a 120 tempos por minuto.

**Andamentos Rápidos:**

Allegro, Vivace e Presto de 120 a 208 tempos por minuto.

Existem, ainda, algumas notações utilizadas para acelerar ou retardar o andamento, ao longo de uma música. Duas delas são: *accelerando* (para tocar mais rapidamente) e *rallentando* (para tocar mais lentamente).

## VAMOS CONSTRUIR UM METRÔNOMO?

Digite o programa da figura 4.12. Ele utiliza uma das características do comando `PLAY`, que é a de dispor de um parâmetro para o metrônomo interno do MSX.

Alterando o valor de `T`, que pode ser qualquer valor, de 32 a 255, você obterá o efeito de um metrônomo eletrônico atual.

Figura 4.12

```
100 KEY(1) ON
110 ON KEY GOSUB 520
120 SCREEN 0,,0:KEY OFF
130 PLAY"s0m200"
140 LOCATE 6,6
150 PRINT"!!! METRONOMO - MSX !!!"
160 LOCATE 2,10
170 INPUT "QUAL O VALOR DE T";T$
180 T=VAL(T$)
190 IF T<32 OR T>255 THEN 160
200 PLAY "T"+T$
210 GOSUB 240
220 PLAY "C4"
230 GOTO 220
240 FOR I= 2312 TO 2319
250 READ A:VPOKE I,A
260 NEXT I
270 LOCATE 2,14:PRINT"(!=";T$;" ) :";
280 DATA 16,16,16,16,16,112,240,224
290 DIM LI(13),A$(13),LS(13)
300 FOR I=0 TO 13
310 READ LI(I),A$(I),LS(I)
320 IF T>=LI(I) AND T<=LS(I) THEN 340
330 NEXT I
340 PRINT A$(I)
350 LOCATE 2,18
360 PRINT"P/MUDAR PRESSIONE F1:";
370 DATA 32,GRAVE LENTO,37
380 DATA 38,GRAVE ,44
```

```

390 DATA 45,LARGO ,51
400 DATA 52,LENTO ,58
410 DATA 59,ADAGIO ,65
420 DATA 66,LARGHETTO ,72
430 DATA 73,MODERATO ,84
440 DATA 85,ANDANTE ,96
450 DATA 97,ANDANTINO ,108
460 DATA 109,ALLEGRETTO ,120
470 DATA 121,ALLEGRO ,149
480 DATA 150,VIVACE ,178
490 DATA 179,PRESTO ,208
500 DATA 209,PRESTISSIMO ,255
510 RETURN
520 PRINT"! AGUARDE !"
530 IF PLAY(1)=-1 THEN 530
540 RUN

```

## RITMO

A palavra ritmo é muito utilizada no contexto musical e às vezes engloba outros conceitos como estilo e divisão (do compasso). É muito difícil definir em palavras o que é ritmo mas, certamente não erraremos se dissermos que:

MÚSICA = SOM + RITMO. onde o ritmo é a repetição de de certos sons, de forma a definir um andamento.

Vejamos, no samba, o ritmo pode ser definido pelo surdo e pela caixa. O tamborim e o pandeiro também entram, fazendo um "desenho rítmico" que caracteriza o andamento e o estilo.


Os instrumentos musicais num conjunto ou numa orquestra podem se revezar, intercalando entre si a condução do ritmo.

Como assimilado o ritmo de uma música pelos ouvintes? De várias formas. Ritmos bastante percussivos podem nos induzir à dança, outros, como o dedilhar de um violão, a bater palmas para "acompanhar a música". Talvez este seja o motivo de alguns pesquisadores definirem o ritmo como sendo som em movimento.

O ritmo não está apenas relacionado com o tipo de compasso, mas também com o andamento. Como será, então, a representação para os instrumentos de percussão? Os ouvidos bem treinados conseguem afinar

um bumbo, um surdo, uma caixa, mas nunca um triângulo pois este tem certas características físicas muito especiais. Assim, veja na figura 4.13 alguns dos símbolos utilizados para a bateria por exemplo.

Figura 4.13



A figura 4.14 mostra a transcrição, para o MSX, da partitura usada como exemplo.

Figura 4.14

```
100 PLAY "s0m400T90o2","s0m400T90o4","s0
m400T90o6"
110 B$="e8.d16e8.d16e8.d16e8.d16"
120 C$="b16r16b16r16b16b16r16b16r16b16b1
6r16b16b16r16b16"
130 P$="r16g16r16g16r16r16g16r16g16r16r1
6g16r16r16g16r16"
140 FORI=1TO3:PLAY B$:NEXT
150 FORI=1TO3:PLAY B$,C$:NEXT
160 PLAY B$,C$,P$
210 GOTO 160
```

## SINAIS DE REPETIÇÃO

Com o desenvolvimento do sistema de notação musical, outros sinais surgiram com o tempo, com a finalidade de facilitar o entendimento e a escrita. Esses sinais têm várias finalidades e são um padrão, como podemos observar nas figuras a seguir:

Figura 4.15 - Sinal de fim de música.





Figura 4.16 - Sinal de divisão de compasso:



Figura 4.17 - Sinal de divisão de período (esse sinal indica o fim de um trecho da música):



Figura 4.18 - Sinal de repetição (ritornello). O número  $n$  representa quantas vezes o trecho é repetido; se nada for indicado, o trecho é repetido apenas uma vez:




Figura 4.19 - Sinal de repetição (casa 1, casa 2, etc.). Deve-se executar a primeira vez, retornar ao início, ou a um ritornello, e, na segunda vez, executar apenas o trecho indicado, não tocando, portanto, o trecho assinalado por "1ª vez":




Figura 4.20 - Sinal de retorno ao início (da capo):

D.C.

Figura 4.21 - Sinais de retorno (normalmente seguidos de outros similares). Assim o executante sabe exatamente o ponto onde deve retornar):



Há ainda os sinais internos aos compassos, como:

Figura 4.22 - Repetição de meio (deve-se repetir o compasso anterior).



Figura 4.23 - Repetição de desenho (deve-se repetir o desenho melódico do compasso indicado, modificando a tonalidade, se indicado).



Figura 4.24 - Abreviação de pausas: (para evitar a repetição de pausas, no arranjo para várias vozes).



Pois, então, munidos da bastante informação, veremos, na próxima aula, como a música ocidental foi estruturada, ao longo da História...

## EXERCÍCIOS

E 4.1- Carregue o programa da figura 4.1 e, após listá-lo, acrescente as linhas mostradas na figura 4.25.

Figura 4.25

```
285 GOTO 500
500 DIM Y(10)
510 FOR I=1 TO 10
520 Y(I)=INT(RND(-TIME)*6)+2
530 IF Y(I)=6 THEN GOTO 520
540 NEXT I
550 FOR I=2 TO 7
560 IF I=6 THEN GOTO 580
570 PRINT I;" ":"P$(I):PRINT
580 NEXT I
590 X=0:U=1
600 FOR N=1 TO 10
602 LOCATE 3,14:PLAY"o6b2o4"
```


```

604 PRINT N;"IDENTIFIQUE O COMPASSO !"
606 LOCATE 3,16:PRINT SPACE$(24)
608 LOCATE 3,18:PRINT SPACE$(24)
610 FOR I=1 TO 4
620 PLAY L$(Y(N))
630 NEXT I
640 LOCATE 3,16:IF PLAY(1) THEN 640
650 INPUT "QUAL A ALTERNATIVA";R$
660 IF VAL(R$)=Y(N) THEN GOSUB 700
670 IF VAL(R$)<>Y(N) THEN GOSUB 800
680 NEXT N
690 LOCATE 3,20:PRINT"SUA NOTA=";X
695 END
700 LOCATE 3,18
710 PRINT "CERTO!";:X=X+U
720 FOR V=0 TO 400:NEXT:U=1:RETURN 680
800 LOCATE 3,18
810 IF U=.5 THEN GOTO 830
815 PRINT "TENDE OUTRA VEZ!"
820 FORV=0 TO 400:NEXT:U=.5:RETURN 602
830 PRINT "ERRADO!"
840 FORV=0 TO 400:NEXT:U=1:RETURN 680

```

Ao rodar este programa, ele mostrará os 5 compassos clássicos na tela, e apresentará 10 testes. Quando aparecer a mensagem "QUAL A ALTERNATIVA?", você deve digitar o número do compasso que o computador tocou algumas vezes e, a seguir, deverá pressionar o RETURN. Acertando na primeira chance, você ganhará 1 ponto. Se você acertar na segunda (e última) chance, ganhará 1/2 ponto.

Figura 4.26



Ao terminar os 10 testes, o MSX dará sua nota.

Treine bastante até conseguir nota 10 em pelo menos 3 baterias seguidas (para começar cada bateria basta digitar RUN e RETURN).


E 4.2- Se você quiser que o compasso seja tocado mais vezes (para tornar os testes mais fáceis) ou menos ágeis (para não dar tanta "colher de chá") que linha da listagem da figura 4.25 você alteraria?

Dica: do jeito que está, cada compasso é tocado 4 vezes

E 4.3- Se você quiser que o programa 4.25 mostre 20 testes, no lugar de 10, que linhas deverá alterar?

E 4.4- Carregue, agora, os programas da figura 4.10, onde você fez a transcrição dos 6 primeiros compassos d'"O GUARANI". Na partitura original está marcado:

Figura 4.27



(que, no nosso metrônomo MSX, corresponderia a um LARGHETTO).

A linha 5, deve ser alterada para

5 PLAY " 50 m500 TXXX "

Qual o valor de XXX que você vai digitar após o T?

E 4.5- Vamos agora, transcrever mais alguns compassos d'"O GUARANI". Veja a Figura 4.28 e transcreva esta partitura para o MSX.

Figura 4.28



Cuidado pois no compasso 9 há uma mudança de andamento: antes de lá você deverá digitar T 140

E 4.6- Grave o programa obtido com a transcrição dos 12 primeiros compassos de "O GUARANI". Veja agora, na figura 4.29 a partitura dos 12 primeiros compassos desta música para a mão esquerda do piano (acompanhamento).

Figura 4.29



Como você pode notar há grupos de 2 notas a serem tocadas simultaneamente

A solução é tocar a mais alta na 2ª voz e a mais baixa na 3ª voz do MSX. Quando a 3ª voz não estiver sendo ocupada, digite uma pausa com a duração do compasso nela (no caso "r 1"). Na figura 4.30 está o complemento do programa com os primeiros 4 compassos completos. Termine a transcrição( dica: coloque o cursor no modo INSERT) e cuidado com a linha 90: o andamento T 140 deve ser colocado nas 3 vozes!

Figura 4.30

```

5  PLAY "s0m5000T120","v10T120","v12T120
10 PLAY "o5e2a4.g16f16","o3g2e2","o3c2o2
a2"
20 PLAY "o5e2a4.g16f16","o3g2e2","o3c2o2
a2"
30 PLAY "e4a4a4o6c4","o3g2e2","o3c2o2a2"

```

40 PLAY "o5b12o6c12o5b12a12g12a12b4g4","  
 o2b12o4c12o2b12a12g12a12b4g4","r1"  
 50 PLAY "o5e2a4.g16f16","??????","??????  
 ??"  
 60 PLAY "e2o4a4.g16f16","??????","??????  
 ??"  
 70 PLAY "e2e4.d8","??????","????????"  
 80 PLAY "e2.g4","??????","??????"  
 90 PLAY "T140a4b4o5c4d4","T140??","????r  
 1"  
 100 PLAY "c4c4c4o4o4b4","????????????????  
 ?????","r1"  
 110 PLAY "a4a4a4g4","?????????????????  
 ?","??"  
 120 PLAY "a2.a4","????????????????????",  
 "??"

E 4.7- Vamos agora transcrever uma partitura com compasso ternário (3/4): a famosa BERCEUSE de BRAHMS.

Figura 4.31





Para digitar mais facilmente, comece digitando uma linha com o aspecto da figura 4.32.

Figura 4.32

```
10 PLAY "
"
"
```

```
"
"
"
```

andando com a barra de espaços para posicionar as aspas e vírgulas (não com as setas!). Comande um LIST para ver se ela tem o aspecto desejado; as três vozes têm seu espaço reservado e estão alinhadas. Leve agora o cursor até o número da linha (10 a 160); uma para cada compasso. Digite as linhas adicionais indicadas na figura 4.33 e complete-a. Antes disso, porém, note que na Clave de Fá temos muitos compassos repetidos. Comande então

```
KEY 2. " c4 g4 g4 " ( e RETURN)
KEY 3. " r4 e4 e4 " ( e RETURN)
KEY 4. " o2 b4 o3 g4 g4" ( e RETURN)
KEY 5. " r4 f4 f4 " ( e RETURN)
```

Desta forma, nos compassos das linhas 10,20,30, 60,90,100,130 e 140, basta posicionar o cursor no começo da 2ª voz, teciar F2, levá-lo até o começo da 3ª voz, teciar F3 e, a seguir, RETURN. A mesma brincadeira pode ser feita com as teclas F4 e F5 nos compassos das linhas 40,50,60, 110 e 150.

Note que a linha 160 tem um compasso de apenas 2 tempos pois ele se completa, na repetição com o pedaço de compasso da linha 8.

Complete agora o programa da figura 4.33.

Figura 4.33

```

5  PLAY "v11T100      "
   "s0m4000T100      "
   "s0m4000T100      "
6  FOR I=1 TO 2
8  PLAY "o4e8e8      "
   "o3r4      "
   "o3r4      "
10 PLAY "g4.e8e4      "
   "c4g4g4      "
   "r4e4e4      "
20 PLAY "g2e8g8      "
   "c4g4g4      "
   "r4e4e4      "
30 PLAY "o5c4o4b4.a8      "
   "c4g4g4      "
   "r4e4e4      "
40 PLAY "a4g4d8e8      "
   "o2b4o3g4g4      "
   "r4f4f4      "
50 PLAY "f4d4d8e8      "
   "o2b4o3g4g4      "
   "r4f4f4      "
60 PLAY "f2d8f8      "
   "o2b4o3g4g4      "
   "r4f4f4      "
70 PLAY "b8a8g4b4      "
   "o3d4o3g4g4      "
   "r4f4f4      "
80 PLAY "o5c2o4c8c8      "
   "??????      "
   "??????      "
90 PLAY "????????      "
   "??????      "
   "??????      "

```



```

100 PLAY "?????"
      "?????"
      "?????"
110 PLAY "?????"
      "?????????"
      "?????"
120 PLAY "?????"
      "?????"
      "?????"
130 PLAY "?????????"
      "?????"
      "?????"
140 PLAY "?????"
      "?????"
      "?????"
150 PLAY "?????????????"
      "?????????"
      "?????"
160 PLAY "c2
      "o3g2
      "o3c2
170 NEXT I

```



E 4.8- Carregue o programa que você gravou quando digitou a figura 4.12. Vamos agora fazer alguns "remendos" nele para dar-lhe outra função. Acrescente as linhas da figura 4.34 .

Figura 4.34

```

90 GOTO 120
115 GOTO 240
155 PRINT:GOTO 240
265 GOTO 290
322 PRINT ")";TAB(13);A$(I);TAB(25);
324 PRINT "(!=";STR$(LS(I));")"
335 GOTO 1000

```

Altere as linhas 140 e 320 como na figura 4.35.

Figura 4.35

```
140 LOCATE 6,2
320 PRINT I;TAB(4);"(!=";STR$(LI(I));
```

e acrescente o trecho listado na figura 4.36.

Figura 4.36

```
1000 X=50
1010 FOR N=1 TO 10
1020 T=INT(RND(-TIME)*223)+32
1030 PLAY "o4T"+STR$(T)
1040 FOR I=1 TO 10
1050 PLAY "c4"
1060 NEXT I
1070 FOR R=0 TO 13
1080 IF T>=LI(R)AND T<=LS(R) THEN 1095
1090 NEXT R
1095 LOCATE 0,20:PRINT STR$(N);
1097 PRINT CHR$(167);" ";
1100 INPUT "QUAL ANDAMENTO";W
1105 LOCATE 22,20:PRINT "-";A$(W)
1110 X=X-(ABS(R-W))
1120 IF PLAY(1)=-1 THEN 1120
1130 LOCATE 3,22:PRINT R;"-";A$(R)
1140 FOR V=0 TO 800:NEXT
1150 LOCATE 3,22:PRINT SPACE$(35)
1155 LOCATE 1,20:PRINT SPACE$(35)
1160 NEXT N
1170 PRINT"SUA NOTA=";X/5
```

Desta forma, seu programa passa a ser um "examinador" que vai submetê-lo a 10 testes, dando uma nota no final.

O "examinador" vai apresentar uma lista de andamentos e começará a tocar o metrônomo.

Em baixo aparecerá o número do teste ( 10,20, etc) e a pergunta:

QUAL O ANDAMENTO?

Você deverá digitar um número escolhido na lista e RETURN.

Após seu número ele colocará o andamento que você escolheu e, quando o metrônomo parar, a resposta correta.

Quanto mais você se afastar da resposta correta, mais pontos perderá. Ao cabo dos 10 testes, você obterá sua nota (que poderá, até ser negativa!).

Rode o programa várias vezes até seu metrônomo mental se ajustar ao do MSX. Será muito difícil você tirar nota 10 (a menos que você seja um cronômetro humano!), mas sua nota deverá ir melhorando com o tempo.


Outra vantagem deste aprendizado é que você irá incorporando a terminologia italiana dos andamentos

JJJ METRÔNOMO - MSX JJJ			
0	(J = 32)	GRAVE LENTO	(J = 37)
1	(J = 38)	GRAVE	(J = 44)
2	(J = 45)	LARGO	(J = 51)
3	(J = 52)	LENTO	(J = 58)
4	(J = 59)	ADAGIO	(J = 65)
5	(J = 66)	LARGHETTO	(J = 72)
6	(J = 73)	MODERATO	(J = 84)
7	(J = 85)	ANDANTE	(J = 96)
8	(J = 97)	ANDANTINO	(J = 108)
9	(J = 109)	ALLEGRETTO	(J = 120)
10	(J = 121)	ALLEGRO	(J = 149)
11	(J = 150)	VIVACE	(J = 178)
12	(J = 179)	PRESTO	(J = 208)
13	(J = 209)	PRESTISSIMO	(J = 255)
12 QUAL ANDAMENTO? ■			

E 4.9- O programa do exercício anterior toca 10 vezes antes de dar a resposta. Que alteração você deverá fazer no programa para que ele toque 20 vezes?


E 4.10-Vamos agora transformar seu MSX numa bateria programável. Como exemplo vamos pegar um ritmo do caribe, o REGGAE. A figura 4.30 mostra a partitura para bateria de um compasso ( que chamaremos Tipo A). Abaixo da partitura está o diagrama de duração das 3 vozes.

Figura 4.38



Vamos usar o esquema de transcrição da figura 4.39.

Figura 4.39



Vamos então digitar um programa que faça esta transcrição Figura 4.40

Figura 4.40

[illegible]

Nas linhas 10 a 20 definimos o envelope, o andamento e as oitava de cada uma das 3 vozes do MSX.


As linhas 30, 40 e 50 definem a string a ser tocada na 1ª, 2ª e 3ª voz do MSX.

Observe com atenção o diagrama das durações e note que os "silêncios" foram preenchidos com pausas de duração adequada.

As linhas de 200 e 230 tocam 5 compassos tipo

A figura 4.41 mostra a partitura e o diagrama de duração para um compasso de REGGAE tipo B.

Figura 4.41




Complete o programa indicado na figura 4.42, digitando corretamente as linhas 60, 70 e 80.

Se você completou corretamente o programa, as linhas de 200 a 230 devem tocar os 10 compassos alternando o A e o B.

Figura 4.42

[illegible]



## AS ESCALAS MUSICAIS

### MODOS

Já comentamos a respeito da influência que a música exerce em nosso estado de espírito, evocando sentimentos que vão da alegria à tristeza. Poderíamos até chegar a extremos, dizendo que ela pode alimentar os processos que nos levam à euforia ou à depressão. Mas, e para atingirmos o equilíbrio?

Se tivermos em mente este objetivo, ou seja, atingirmos o equilíbrio, poderemos perceber claramente os motivos pelos quais não se executam, atualmente, rock heavy-metal ou mesmo um samba dentro de um templo religioso.

A música comercial atual traz mensagens cantadas e tocadas. Mesmo retirando-se a mensagem cantada e executando-se a parte tocada de um samba ou rock, nosso organismo responde à excitação sonora de uma forma que certamente não é o equilíbrio.

Na história do ocidente, a Igreja Cristã representou um elemento que ditou regras para a composição e execução das músicas com o propósito de evocar a Deus e atingir o equilíbrio necessário para agirmos com respeito e coerência.

A documentação histórica representa um papel extremamente importante na busca do caminho percorrido pelos compositores ao longo das eras. Veremos, a seguir, como surgiram as regras que ditam a composição da música.


## MÚSICA, MATEMÁTICA, FILOSOFIA

O estudo das relações matemáticas na música praticamente se inicia com Pitágoras, no século VI a.C., Pitágoras, filósofo grego, admitia como essência do Universo as relações numéricas. Segundo ele, cada planeta vibrava segundo uma tonalidade harmônica (ou um som agradável) e que, conhecendo a relação numérica entre esses sons, poderia atingir o equilíbrio e desvendar grandes segredos do Universo.

Na acrópole de Atenas, na Grécia, era comum aos homens até 30 anos, o estudo da música e dos instrumentos musicais, em particular, a flauta. A música era composta e executada segundo sons discretos e pertencentes a uma extensão bem definida, talvez limitada pelos instrumentos musicais ou até mesmo pela teoria. Basicamente já se utilizava praticamente a escala diatônica atual.

Lá pelo século II d.C., Ptolomeu, outro filósofo grego, acabou por definir as idéias de Pitágoras e estabelecer novas relações matemáticas mais precisas, adotando a forma da figura 5.1 como a escala verdadeira, também chamada escala de Ptolomeu, que é praticamente a mesma de Pitágoras.

Figura 5.1 - Representação atual da escala de Ptolomeu



O desenvolvimento da música medieval que, na história do ocidente, vai, digamos, até o fim do século XIII, foi baseado nesta mesma sucessão de sons, embora com outras extensões.


Essas sequências, chamadas modos, são precursoras das escalas atuais e estabeleciam as regras das composições. Todas as notas escolhidas para uma dada sequência de notas, ou melodia, deveriam pertencer a um único modo e a primeira e última notas determinavam o nome do modo utilizado.

Esses modos, pela sua origem grega, tiveram seus nomes escolhidos a partir de estilos arquitetônicos usados na Grécia. A figura 5.2 mostra o templo do Partenon, em Atenas, projetado pelos arquitetos Ictinos e Calícrates e construído todo em mármore, no estilo dórico. Outro nome usado, o jônico, vem da Jônia, na Ásia Menor, de onde era originário o poeta



Homero e definia um estilo arquitetônico mais rebuscado.

Figura 5.2 - Partenon, em Atenas.



Santo Ambrósio (séc. IV d.C.), bispo de Milão, adotou os modos, dórico, frígio, lídio e mixolídio chamados autênticos, mostrados na figura 5.3.

Figura 5.3 - Modos autênticos.




Os modos autênticos estão representados na clave de Sol, com uma altura bem definida mas, na verdade, eram transpostos para a altura da voz para a qual a melodia era escrita, normalmente para tenores, ou seja, na Clave da Fá.


Para possibilitar novas técnicas de composição para duas vozes, o chamado "organum" paralelo, São Gregório (séc. VI d.C.) adotou outros modos, derivados destes quatro, os modos plagais. As vozes mais graves cantavam, então, no modo plagal. As quatro notas mais agudas de cada modo foram transportadas para a região mais grave (figura 5.4).

Figura 5.4 - Modos plagais.


modo hipodórico



modo hipofrígio



modo hipolídio



modo hipomixolídio



Digite o programa da figura 5.5 para ouvir os quatro modos descritos e seus respectivos plagais.

Figura 5.5 - Modos musicais.

```

100 SCREEN 0,,0:KEY OFF
110 PRINT "MODO DORICO"
120 PLAY"v15o4defgabo5cd"
130 GOSUB 2000
140 PRINT "MODO FRIGIO"
150 PLAY"v15o4efgabo5cde"

```

```

160 GOSUB 2000
170 PRINT "MODO LIDIO"
180 PLAY"v15o4fgabo5cdef"
190 GOSUB 2000
200 PRINT "MODO MIXOLIDIO"
210 PLAY"v15o4gabo5cdefg"
220 GOSUB 2000
230 PRINT "MODO HIPODORICO"
240 PLAY"v15o3abo4cdefga"
250 GOSUB 2000
260 PRINT "MODO HIPOFRIGIO"
270 PLAY"v15o3bo4cdefgab"
280 GOSUB 2000
290 PRINT "MODO HIPOLIDIO"
300 PLAY"v15o4cdefgabo5c"
310 GOSUB 2000
320 PRINT "MODO HIPOMIXOLIDIO"
330 PLAY"v15o4defgabo5cd"
340 GOSUB 2000
350 END
2000 IF PLAY(1) THEN 2000
2100 FOR T=0 TO 800:NEXT:RETURN

```

Seguiram-se outros três modos, devido à pouca flexibilidade dos já existentes. Veja a figura 5.6 .

Figura 5.6 - Modos jônico, aeólio e iócrio.



A primeira vista tudo pode parecer até redundante e estranho, mas, conforme veremos, os modos jônico e aeólio persistem até hoje, com os nomes de escala maior e escala menor.

Vamos executar esses modos?

A figura 2.11 (capítulo 2) apresenta uma escala contínua que inclui todos esses modos. Se você obedecer às regras de composição já citadas, perceberá que a criação de melodias dentro desses modos é bastante restrita. Contudo, as denominações tônica e dominante, muito utilizadas vêm dos modos autênticos e plagais. A tônica é a primeira (ou última) nota de modo autêntico e a dominante é a nota que aparece com mais frequência nas melodias escritas no modo plagal.

## AS TECLAS PRETAS DO PIANO

Você já deve ter se questionado a respeito dessa característica do piano, não é? Bem, o piano, chamado na época de "pianoforte", foi inventado por volta de 1700 por Bartolomeo Cristofori, um italiano. Ele o chamou, inicialmente de "clavicembalo col piano e forte", ou seja, cravo com suave e forte. O cravo, seu predecessor, é derivado do primeiro instrumento de teclas e cordas, o virginal, criado no século XV. Todos são teoricamente parecidos. Mas e o caso das teclas pretas? Observe a parte superior da figura 2.11.

Figura 5.7



O som das teclas pretas surgiu num processo confuso entre os séculos XIII e XVI. Esses sons que não havíamos comentado até agora, são outros indicadores de altura, de um processo chamado "temperamento". Vejamos como eles surgiram.

O canto gregoriano (lembra-se de São Gregório?), evoluiu para formas mais complexas, para corais de várias vozes, cada uma cantando uma melodia. O entrelaçamento dessas vozes numa mesma música, é um processo chamado modulação (mudança de modo) que podia produzir desarmonia no som resultante. Os executantes, quando chegavam nessas situações, modificavam ligeiramente a altura de alguns sons e obtinham combinações


mais agradáveis ao ouvido. Para não violar as rígidas regras do canto gregoriano, nenhuma indicação era feita nas partituras, mas um dos primeiros efeitos utilizados foi o de diminuir ligeiramente a altura da nota Si, na modulação do modo jônico para o lídio. Vejamos o porque.

## TOM E SEMITOM

Uma das características descobertas por Pitágoras e Ptolomeu foi a estrutura de intervalos da escala verdadeira. Intervalo, como o nome diz é a distância sonora entre dois sons, ou a diferença entre as frequências de dois sons, como veremos nos capítulos 9 e 10.


Utilizando o programa da figura 2.11 toque sucessivamente no MSX as sequências mostradas na figura 5.8 .

Figura 5.8 - Sequências de sons.




Tente agora as sequências da figura 5.9.

Figura 5.9 - Sequências de sons.



Você deve ter percebido que as sucessões Mi-Fá e Si-Dó parecem mais próximas que Dó-Ré e Fá-Sol. Pois é, conforme Ptolomeu, na escala diatônica temos as distâncias da figura 5.10 .


Figura 5.10



Chamamos atualmente a essas distâncias longas e curtas, tom e semi-tom (ou meio-tom), respectivamente. Nos modos eclesiásticos (assim chamados por causa da sua adoção pela igreja), essa estrutura varia de modo para modo, como se pode perceber, já que cada modo tem por tônica uma nota diferente.

Vejamos o efeito da modulação e o que aconteceria se executássemos uma mesma melodia em dois modos distintos, como na figura 5.11

Figura 5.11 - Execução de uma melodia em dois modos.



Para ouvir ambas separadamente e simultaneamente, digite o programa 5.12.

Figura 5.12

```

10 SCREEN 0,,0:KEY OFF
20 PLAY "s0m5000o3","s0m5000o3"
30 A$="L4ceeeefedc2."
40 B$="L4faafaabagf2."
60 PRINT"voz A":PLAY A$:GOSUB 120
70 PRINT"voz B":PLAY B$:GOSUB 120
90 PRINT"voz A+B":PLAY A$,B$:GOSUB 120
110 END
120 IF PLAY(1) THEN 120
130 FOR T=0 TO 500:NEXT:RETURN

```

NOTA : A representação foi feita propositalmente na oitava 3 para soar mais nitidamente ao ouvido.

Se você substituir a nota Si (B) por Si bemol (B-), perceberá o efeito do abaixamento na altura da nota Si, o que trará maior harmonia no soar da melodia. Acrescente ao seu programa as linhas 50, 80 e 100

para torná-lo igual ao da figura 5.13 e rode-o novamente.

Figura 5.13

```
10 SCREEN 0,,0:KEY OFF
20 PLAY "s0m5000o3","s0m5000o3"
30 A$="L4ceeeceefedc2."
40 B$="L4faafaabagf2."
50 BB$="L4faafaab-agf2."
60 PRINT"voz A":PLAY A$:GOSUB 120
70 PRINT"voz B":PLAY B$:GOSUB 120
80 PRINT"voz B bemolizada":PLAY BB$:GOSUB 120
90 PRINT"voz A+B":PLAY A$,B$:GOSUB 120
100 PRINT"voz A+B bemolizada":PLAY A$,BB$:GOSUB 120
110 END
120 IF PLAY(1) THEN 120
130 FOR T=0 TO 500:NEXT:RETURN
```

Essa diferença que você notou na modulação de Dó para Fá, ou seja, na mudança de modos, com a nota Si original é a desarmonia de que falamos. Os intérpretes, a gosto, "bemolizavam" a nota Si em construções deste tipo.

Pelos estudos de Ptolomeu, o problema reside no intervalo entre as notas Mi-Fá e Lá-Si. O primeiro é de um semi-tom. O segundo é de um tom. O abaixamento da altura da nota Si acabou gerando a nota Si bemol, que atualmente corresponde a uma das teclas pretas do piano.

Mas e as outras teclas pretas? É o temperamento, o que é? Bem, esses dois assuntos veremos no capítulo seguinte.

**NOTA :** Como você já percebeu, estamos "passeando" pelo planeta através de séculos e milênios. Os fatos que mencionamos são apenas estímulos para que sua curiosidade o faça pesquisar em bons livros de história da música e de história geral alguns fatos importantes. As manifestações artísticas refletem, ao longo dos séculos, a evolução de cada sociedade e toda uma realidade econômica, política e social. Quem compreenderia nosso tipo de música, numa outra região e numa outra época se não conhecesse nossos envolvimento com o mundo, nossos conhecimentos, nossa tecnologia? Os ar-

tistas, dizem, são pessoas muito sensíveis que conseguem se expressar muito melhor, através de sua arte.

## EXERCÍCIOS

E 5.1- Béla Bartók (1881-1945) foi um compositor húngaro, autor de várias obras primas como "Música para cordas, percussão e celesta" (1936) e "Concerto para orquestra" (1943). Lecionou piano na Academia de Música de Budapest e escreveu, para seus alunos, algumas peças simples. A figura 5.14 mostra a partitura da peça "EM MODO DÓRICO".

Figura 5.14 - Peça "EM MODO DÓRICO".

Lento, J. 194

*p, legato*

The musical score is written for piano and consists of four systems of two staves each. The tempo is 'Lento' and the time signature is 'J. 194'. The dynamics are 'p, legato'. The key signature is one sharp (F#). The piece ends with a double bar line and a key signature change to one flat (F).

[A2 194]



Com base nessa partitura, complete o programa da figura 5.15. Lembre-se dos truques que lhe foram ensinados para economizar tempo e digitação.

Figura 5.15

```

5  PLAY "som5000", "som5000", "som5000"
6  PLAY "T120", "T120", "T120"
10 PLAY "r4o5d4c4o4b4a2",
20 PLAY "a4b4o5c4o4g4a2",
30 PLAY "a4g4a4b4o5c4d4",
40 PLAY "o4b1bgo5c4",
50 PLAY "o4",
60 PLAY "",
70 PLAY "",
80 PLAY "",
90 PLAY "",
100 PLAY "",
110 PLAY "",
120 PLAY "",
130 PLAY "",
140 PLAY "1".

```



Apesar da peça estar totalmente escrita em modo dórico (como o próprio nome diz) apenas no acorde final há necessidade de um sinal especial (\*). Mas não se preocupe: veremos isso na próxima aula. Na listagem o sinal correspondente já foi colocado no último compasso da linha 140.

E 5.2- Outra peça que Béla Bartók escreveu para seus alunos chama-se "EM MODO FRÍGIO" e sua partitura está na figura 5.16.

Figura 5.16

Calmo, *al. se*

*p, legato*


*mf*

*sf*

*dim.*

*p*

[et...]



Digite o programa da figura 5.17 e complete-o com base na partitura dada.

Figura 5.17

```

4  PLAY "s0m20000" "s0m20000"
5  PLAY "T80" "T80"
6  SCREEN 0:KEY OFF:WIDTH 39
7  PRINT"BELA BARTOK:"
8  PRINT"EM MODO FRIGIO"
9  PRINT"CALMO"
10 PLAY "r4o5c4o4b2      "
    "o3e2.f4             "
20 PLAY "b4o5?????      "
    "e?.g?               "
30 PLAY "b?o5?????      "
    "??.?4               "
40 PLAY "?????????      "
    "??                  "
50 PLAY "?????????      "
    "?????????          "
60 PLAY "?????????      "
    "b1                  "
70 PLAY "??              "
    "?????????          "
80 PLAY "??              "
    "?????????          "
90 PLAY "?????????????  "
    "e2b2                "
100 PLAY "f2c2            "
    "?????????          "

```

110	PLAY	"d2r4o4??	"
		"????	"
120	PLAY	"o5??????	"
		"?????	"
130	PLAY	"????????	"
		"??????	"
140	PLAY	"????????	"
		"????	"
150	PLAY	"????????	"
		"????	"
160	PLAY	"o4????????	"
		"????	"
170	PLAY	"????????	"
		"????	"
180	PLAY	"????????????????	"
		"????	"
190	PLAY	"o4??????????	"
		"?i	"
200	PLAY	"e??????????	"
		"????????	"
210	PLAY	"????????????	"
		"b1	"
220	PLAY	"????????????	"
		"b4a4g4a4	"
230	PLAY	"?????	"
		"f2b2	"
240	PLAY	"b1	"
		"????	"
250	PLAY	"b1	"
		"e1	"
260	PLAY	"b1	"
		"e1	"



E 5.3- A famosa cantiga do "pastorzinho" tem uma letra, no refrão, que é a própria partitura:

Dó	Ré	Mi	Fá	.....	Fá	Fá	...
Dó	Ré	Dó	Ré	.....	Ré	Ré	...
Dó	Sol	Fá	Mi	.....	Mi	Mi	...
Dó	Ré	Mi	Fá	.....	Fá	Fá	...

Digite o programa da figura 5.18 para fazer seu MSX tocar esta cantiga.

Figura 5.18 -


```

5  PLAY "s0m5000L4T200o4"
10 PLAY "cdefrffr"
20 PLAY "cdcdrrdr"
30 PLAY "cgfereer"
40 PLAY "cdefrffr"

```

Digamos, agora, que você queira tocar a cantiga num modo mais grave, deslocando a música 3 notas para baixo. Para isso basta fazer a correspondência da figura 5.19.

Figura 5.19



a cantiga, transposta em 3 notas para baixo, ficará:

Dó Ré Mi Fá .... Fá Fá ...  
(Sol) (Lá) (Si) (Dó) (Dó) (Dó)

Dó Ré Dó Ré .... Ré Ré ...  
(Sol) (Lá) (Sol) (Lá) (Lá) (Lá)

Dó Sol Fá Mi .... Mi Mi ...  
(Sol) (Ré) (Dó) (Si) (Si) (Si)

Dó Ré Mi Fá .... Fá Fá ...  
(Sol) (Lá) (Si) (Dó) (Dó) (Dó)

Acrescente as linhas 50, 60, 70 e 80 ao programa anterior, de maneira que ele toque a cantiga segundo esta nova sequência. Cuidado que as notas usadas agora, com exceção do Dó e Ré, pertencem à oitava 3.

Figura 5.20

```

5  PLAY "s0m5000L4T200o4"
10 PLAY "cdefrffr"
20 PLAY "cdcdrrddr"
30 PLAY "cgfereer"
40 PLAY "cdefrffr"
50 PLAY "o3gabo4crccr"
60 PLAY "o3????r??r"
70 PLAY "?o4??o3?r??r"
80 PLAY "?????????????"

```

E 5.4- Digamos, agora, que você queira tocar a cantiga original começando 3 notas acima do Dó (e não três abaixo como fizemos na figura 5.20). A correspondência, neste caso, é dada pela figura 5.21. A cantiga assim transposta, ficará:



<div style="display: flex; align-items: center;"> <div style="margin-right: 5px;">3</div> <div style="border-left: 1px solid black; padding-left: 5px;"> <div style="margin-bottom: 5px;">SI</div> <div style="margin-bottom: 5px;">LÁ</div> <div style="margin-bottom: 5px;">SOL</div> <div style="margin-bottom: 5px;">FÁ</div> <div style="margin-bottom: 5px;">MI</div> <div style="margin-bottom: 5px;">RÉ</div> <div style="margin-bottom: 5px;">Dó</div> </div> </div>	<div style="margin-bottom: 5px;">MI</div> <div style="margin-bottom: 5px;">RÉ</div> <div style="margin-bottom: 5px;">Dó</div> <div style="margin-bottom: 5px;">SI</div> <div style="margin-bottom: 5px;">LÁ</div> <div style="margin-bottom: 5px;">SOL</div> <div style="margin-bottom: 5px;">FÁ</div> <div style="margin-bottom: 5px;">...</div>
---	---

Dó	Ré	MI	Fá ....	Fá	Fá ...
(Fá)	(Sol)	(Lá)	(Si) ....	(Si)	(Si)

Dó	Ré	Dó	Ré ....	Ré	Ré ...
(Fá)	(Sol)	(Fá)	(Sol) ...	(Sol)	(Sol)

Dó	Sol	Fá	MI ....	MI	MI ...
(Fá)	(Dó*)	(Si)	(Lá) ....	(Lá)	(Lá)

Dó	Ré	MI	Fá ....	Fá	Fá ...
(Fá)	(Sol)	(Lá)	(Si) ....	(Si)	(Si)

\* ( da oitava 5)

Complete, então, o programa da figura 5.22 . Com as linhas 90, 100, 110 e 120 e ouça sua melodia transposta.

Figura 5.22

```
5  PLAY "s0m5000L4T200o4"  
10 PLAY "cdefrffr"  
20 PLAY "cdcdrrdr"  
30 PLAY "cgfereer"  
40 PLAY "cdefrffr"  
50 PLAY "o3gabo4crccr"  
60 PLAY "o3gagaraar"  
70 PLAY "go4dco3brbbr"  
80 PLAY "o3gabo4crccr"  
90 PLAY "fgabrbbbr"  
100 PLAY "????r??r"  
110 PLAY "?o5co4b?r??r"  
120 PLAY "????????"
```

Se quiser, vá conferir a resposta correndo, mas não se preocupe, a horrível "desafinação" que o último trecho produziu não é culpa sua. Veja o exercício seguinte.

E 5.5- Acontece que, ao tentar tocar a melodia num modo mais agudo, nos defrontamos com o problema de obter um intervalo de um tom quando precisamos de apenas meio-tom. Quando tocamos

Dó Ré Mi Fá ..... Fá Fá ...

na cantiga original, entre o Mi e o Fá, tínhamos um intervalo de um semi-tom. Ao transportarmos para

Fá Sol Lá Si .... Si Si ...

usamos, entre o Lá e o Si, um intervalo de um tom! Para consertar isso, devemos "abaixar" o Si (b), bemolizando-o. No MSX, basta digitar um sinal de menos (-) logo após a cifra do Si (b-).

Conserte as linhas 90, 110 e 120 do programa anterior e rode-o para sentir (com "alívio") a diferença!

Figura 5.23

```
5  PLAY "s0m5000L4T200o4"  
10 PLAY "cdefrffr"  
20 PLAY "cdcdrrddr"  
30 PLAY "cgferer"  
40 PLAY "cdefrffr"  
50 PLAY "o3gabo4crccr"  
60 PLAY "o3gagaraar"  
70 PLAY "go4dco3brbbr"  
80 PLAY "o3gabo4crccr"  
90 PLAY "fgab-rb-b-r"  
100 PLAY "fgfgrggr"  
110 PLAY "fo5co4b?araar"  
120 PLAY "fgab?rb?b?r"
```







## OS ACIDENTES MUSICAIS


### OS ACIDENTES

Neste capítulo observaremos os processos que levam ao aparecimento desses novos indicadores de altura, os acidentes, já vimos que, historicamente, certas notas acidentais, para soar melhor ao ouvido, tinham sua altura modificada pelos executantes. Vejamos como e quando ocorre.


Tomemos como exemplo o modo dórico. Vamos supor que estamos executando uma melodia nesse modo, iniciando-a e terminando-a pela nota Ré. Se executarmos essa mesma melodia, partindo da nota Dó, no modo jônico, obteremos um resultado final que soa de modo diferente, pois a estrutura dos dois modos é diferente (veja a figura 6.1).

Figura 6.1- T- TOM

S- SEMI-TOM



modo Dórico



modo Jônico

Digite o programa da figura 6.2 para ouvi-los no MSX.

Figura 6.2

```
10 PLAY "s0m5000L4o4"  
20 PRINT:PRINT "MODQ JONICO"  
30 PLAY "ccdefgabo5ccrco4bagfedc"  
40 IF PLAY(1) THEN 40  
50 PRINT:PRINT "MODO DORICO"  
60 PLAY "rdefgabo5cdrdco4bagfed"  
70 IF PLAY(1) THEN 70  
80 GOTO 20
```


A linha 30 representa o modo jônico e a linha 60, o modo dórico. A diferença entre as duas escalas é devida à estrutura dos intervalos entre as notas, mostrada na figura 6.1. Modifique as linhas 50 e 60 para o indicado na figura 6.3.

Figura 6.3

```
50 PRINT:PRINT "MODO DORICO ALTERADO"  
60 PLAY "rdef#gabo5c#drdc#o4bagf#ed"
```

e ouça novamente... Ambas as escalas soam de forma bastante semelhante. A solução dada foi aumentar as alturas das notas Fá e Dó para obtermos a mesma estrutura do modo jônico. Esta é a chamada escala de RÉ MAIOR. Na notação musical, representamos esses aumento na altura, de meio tom, pelo símbolo #, chamado sustenido, que quer dizer elevado. A estrutura da escala de RÉ MAIOR é mostrada na figura 6.4.

Figura 6.4 - Escala de RÉ MAIOR.




ESCALA DE RÉ MAIOR


Se repetirmos o processo para a escala que começa pela nota Mi, teremos o resultado mostrado na figura 6.5 .

Figura 6.5

ESCALA DE MI  
(modo Frígio)



ESCALA DE MI MAIOR



Altere seu programa como indicado na figura 6.6 .

Figura 6.6

```

10 PLAY "s0m5000L4o4"
20 PRINT:PRINT "MODO JONICO"
30 PLAY "rcdefgabo5crco4bagfedc"
40 IF PLAY(1) THEN 40
50 PRINT:PRINT "RE MAIOR"
60 PLAY "rdef#gabo5c#drdc#o4bagf#ed"
70 IF PLAY(1) THEN 70
80 PRINT:PRINT "MI MAIOR"
90 PLAY "ref#g#abo5c#d#ered#c#o4bag#f#e"
100 IF PLAY(1) THEN 100
500 GOTO 20

```

Você perceberá que as escalas maiores soam sempre de forma semelhante porque têm a mesma estrutura.

Vejamos o que ocorre com a escala de Fá. Digite as linhas da figura 6.7 .

Figura 6.7

```

110 PRINT:PRINT "FA"
120 PLAY "rfgabo5cdefrfdco4bagf"
130 IF PLAY(1) THEN 130

```


Parece que a estrutura está incorreta. Se analisarmos a estrutura da escala de FÁ, vamos perceber que, abaixando a altura da nota Si de meio tom, a estrutura "acerta". Para você confirmar, altere as linhas indicadas na figura 6.8.

Figura 6.8


```
110 PRINT:PRINT "FA MAIOR"
120 PLAY "rfgab-o5cdefrfedco4b-agf"
```

Esse efeito, na notação musical, é indicado pelo sinal  $b$ , chamado bemol (figura 6.9).

Figura 6.9



Escala de FÁ



Escala de FÁ MAIOR

Para ouvir as escalas de SOL MAIOR, LÁ MAIOR e SI MAIOR acrescente as linhas mostradas na figura 6.10.

Figura 6.10

```
140 PRINT:PRINT "SOL MAIOR"
150 PLAY "rgabo5cdef#grgf#edco4bag"
160 IF PLAY(1) THEN 160
170 PRINT:PRINT "LA MAIOR"
180 PLAY "rabo5c#def#g#arag#f#edc#o4ba"
190 IF PLAY(1) THEN 190
200 PRINT:PRINT "SI MAIOR"
210 PLAY "rbo5c#d#ef#g#a#brba#g#f#ed#c#o4b"
220 IF PLAY(1) THEN 220
```

NOTA : No comando PLAY, os sustenidos são representados pelo sinal + ou # e os bemóis pelo sinal -.

No processo que desenvolvemos surgiram seis novos indicadores de altura, que são:

Dó#, Ré#, Fá#, Sol#, Lá# e Si

Essas duas últimas notas representam,

Lá# -- meio tom acima da nota Lá.

Sib -- meio tom abaixo da nota Si.

Como o intervalo Lá-Si é de um tom, ambas representam a mesma altura. Os nomes são distintos porque vêm de escalas distintas.

## NOVAS ESCALAS

Você poderá perguntar, "Se há novas notas, como serão as escalas maiores dessas novas notas? Será que aparecerão novas notas?"

Bem, a estrutura da escala diatônica já sugere que não. Se há 5 tons e 2 semitons e se 1 tom divide-se em 2 semitons iguais, só podem existir 12 semitons na escala. O que muda é apenas o nome dado a essas notas.

Voltando à nossa pergunta, construiremos uma nova escala maior, partindo da nota DO#. A resposta está no programa modificado pelas linhas da figura 6.11.


Figura 6.11

```
230 PRINT:PRINT "DO# MAIOR"  
240 PLAY "rc#d#e#f#g#a#o5b#c#rc#b#o4a#g#  
f#e#d#c#"   
250 IF PLAY(1) THEN 250  
500 GOTO 20
```

Note que, no MSX, o B# já cai na oitava 5 pois o limite da oitava 4 é o B. Cuidado com isso quando for transcrever partituras!

Nessa escala de DO# maior, (figura 6.12) todas as notas são aumentadas de meio tom. A altura da nota Mi# é a mesma da nota Fá e a da nota Si# é a mesma da nota Dó.

Figura 6.12 - Escala de Dó# MAIOR.



Se prosseguirmos, chegaremos a situações onde uma nota deverá receber dois sustenidos. Para essa situação utilizamos o dobrado sustenido (figura 6.13).

Figura 6.13 - O Símbolo do dobrado sustenido



O mesmo pode ocorrer em certas escalas, com dois bemóis e o símbolo usado é o dobrado bemol (figura 6.14).

Figura 6.14 - O Símbolo do dobrado bemol.



A partir daí, o processo se esgota pois, um dobrado sustenido ou um dobrado bemol representam a maior distância sonora entre notas consecutivas da escala diatônica. Adotamos, então, o nome dessa nota na escala. Por exemplo, a escala de FÁ DOBRADO SUSTENIDO é a escala de SOL.

Você já percebeu que essas escalas maiores podem ser escritas em qualquer oitava. Os mesmos acidentes aparecem em qualquer oitava que você tocar uma escala maior, (exceto na escala de DÓ maior que não tem acidentes).

É por isso que, quando uma música é baseada numa escala maior, os acidentes das notas dessa escala, quando ocorrem, são representados junto à clave e não no decorrer da música. Veja um exemplo de início de partitura na figura 6.15.

Figura 6.15 - Peacherino Rag : Scott Joplin (1901)



## ALGUMAS CURIOSIDADES

Todas as redundâncias dos vários nomes para uma mesma altura de som, como vimos para as notas **LÁ** e **SÍ** são devidas à estrutura da escala diatônica. Há, contudo, um porém... Na verdade Pitágoras e Ptolomeu adotaram uma estrutura diferente daquela que adotamos. A relação entre os sons da escala verdadeira foi definido como na figura 6.16.

Figura 6.16 - Escala verdadeira.



Se adotarmos essa hipótese, nossas deduções serão falsas, pois, se iniciarmos as escalas por outras notas, teremos que adotar outros indicadores de altura e chegaremos à conclusão que **LÁ SUSTENIDO** é diferente de **SÍ BEMOL**.

Fisicamente a escala de Ptolomeu é realmente a verdadeira. Até o século XVII havia uma grande restrição para execução das músicas pois os instrumentistas "afinavam" seus instrumentos para músicas compostas numa determinada escala. A diferença entre o tom maior (T) e o tom menor (t) quase não é percebida pelo ouvido da maioria das pessoas.

Figura 6.17




JOHANN SEBASTIAN BACH

Johann Sebastian Bach, compositor alemão nascido em 1685, adotou a estrutura baseada na igualdade de T e t, na afinação dos instrumentos, chamando a esse método de temperamento. O sistema temperado é que define a estrutura que estamos estudando e divide a escala em doze semitons iguais, sendo um tom igual a dois semitons. A diferença de altura entre o som temperado e o som real é chamada coma.

O instrumento de teclado mais utilizado na época, derivado de um instrumento mais antigo, chamado virginal, era o cravo. Bach afinou seu cravo no sistema temperado e compôs uma série de estudos para esse instrumento, chamando sua obra de "O Cravo bem temperado".

As peças para cravo e orquestra forçaram os outros instrumentistas a tocar também no sistema temperado. Portanto, vemos agora desvendar-se o mistério das teclas pretas do piano, como mostra a figura 6.18.

Figura 6.18 - Teclado de um piano.




A maioria dos instrumentos de teclado é afinada no sistema temperado e o MSX também toca, através do comando PLAY, na escala temperada.



## O MSX "BEM TEMPERADO"

O programa da figura 6.20 transforma uma parte do teclado do seu MSX num teclado de piano. A correspondência entre os dois teclados está mostrada na figura 6.19.

Figura 6.19 - Correspondência dos teclados.



O timbre do MSX é diferente do timbre do piano, mas assim mesmo você poderá executar muitas músicas com ele.

Figura 6.20

```
10 POKE &HFCAB,255
20 DIM N(255):PLAY"=0m3000LB"
30 FOR I=0 TO 19
40 READ D$:D=ASC(D$):N(D)=I+30
50 NEXT I
60 DATA A,Z,S,X,D,C,V,G,B,H,N,M,K,"","L,
  ",",/,*
65 GOSUB 1000
70 A$=INPUT$(1):A=ASC(A$):N$="N"+STR$(N(
A))
80 IF N(A)=0 THEN GOTO 70 ELSE PLAY N$
90 GOTO 70
```

```


1000 OPEN"GRP:"AS #1
1010 COLOR 1,15,15:SCREEN 2
1020 T$="C1S4R10E2D14H2L10G2U14F2D10G2R1
4H2U10E2L14"
1030 FOR I= 1 TO 12
1040 READ A$
1050 PRESET(20*I,100):DRAW T$
1060 PRESET(20*I+3,102):PRINT#1,A$
1070 NEXT I
1080 DATA A,S,D,F,G,H,J,K,L,,',*,
1090 FOR I= 1 TO 11
1100 READ A$
1110 PRESET(20*I+10,119):DRAW T$
1120 PRESET(20*I+13,121):PRINT#1,A$
1130 NEXT I
1140 DATA Z,X,C,U,B,N,M,"","'",";","/
1150 T1$="C1L20U48R9U36L18D36R9"/
1160 T2$="C1L20U48BU18U18"
1170 FOR I=0 TO 11
1180 PRESET (45+20*I,162):IF I=3 OR I=6
OR I=10 THEN DRAW T2$ ELSE DRAW T1$:PAIN
T (25+20*I,90),1,1
1190 NEXT I
2000 FOR L=9 TO 41 STEP 8
2010 LINE(0,L)-(255,L)
2020 NEXT L
2030 FOR I=1 TO 11
2040 D=14+20*I
2050 PRESET(D,62-4*I):PRINT#1,CHR$(1)+
CHR$(73)
2060 NEXT I
2070 LINE(33,57)-(41,57)
2072 LINE(33,49)-(41,49)
2074 LINE(53,57)-(61,57)
2076 LINE(53,49)-(61,49)
2078 LINE(73,49)-(81,49)
2088 LINE(93,49)-(101,49)
2110 Z$="C1S12LHUERM+2,+1M+1,+2M-1,+2M-2
,+1M-2,-1H2U2M+1,-2E4U2HD16GH"
2120 PRESET(21,32):DRAW Z$:PRESET(22,33)
:DRAW Z$:PRESET(22,32):DRAW Z$:PRESET(21
,33):DRAW Z$
2130 FOR I=1 TO 11
2140 READ A$
2150 U=8+20*I
2160 FOR L=0 TO LEN(A$)-1
2170 PRESET(U+6*L,154):PRINT#1,MID$(A$,L
+1,1)

```

```

2180 NEXT L,I
2190 DATA sol,lá,si,dó,ré,mi,fá,sol,lá,s
i,dó
3000 RETURN

```

Experimente tocar a cantiga do "pastorzinho" em seu "piano" como indicado nos exercícios 5.3 a 5.5. Note que, na transposição para o modo mais agudo você deverá tocar uma tecla preta (SI BEMOL = ) para "encaixar" a melodia.

Curiosidade: Se você é uma pessoa interessada em inovações, já deve ter visto alguns tecladistas empunhando teclados portáteis e achado uma idéia interessante. Saiba, então, que esse tipo de instrumento data de ao redor de 1800, era chamado ORPHICA e foi projetado na Áustria.

## EXERCÍCIOS

- E 6.1- Toque uma escala de Ré MAIOR, em seu MSX usando o programa da figura 6.20. Qual a sequência de teclas do MSX que você usou?
- E 6.2- Toque uma escala de DO# MAIOR em seu MSX. Qual a sequência de teclas que você usou?
- E 6.3- Usando, ainda, o teclado da figura 6.20, analise a partitura mostrada na figura 6.21.

Figura 6.21 - Oh! Minas Gerais!



Ela corresponde à famosa:

```
" Oh! Minas Gerais!  
Oh! Minas Gerais!  
Quem te conhece  
Não esquece jamais.  
Oh! Minas Gerais"
```

cuja melodia, por sinal, foi trazida de uma música italiana convidando para o mar ( que ironicamente, é a única coisa que Minas Gerais não tem!).

Escreva na partitura, sob cada nota, qual a tecla do seu "MSX bem temperado" que você deve apertar para tocar esta música e depois tente executá-la. Não fique pressionando cada tecla por muito tempo pois você disparará a rotina de repetição, e terá que ficar aguentando uma sequência de notas idênticas!

- E 6.4- Certifique-se de que o programa da figura 6.20 foi bem gravado e ressete o computador. Agora, com base na partitura da figura 6.21, complete o programa da figura 6.22 .

```
Figura 6.22      5   PLAY "som9000T138o4"  
                  6   GOSUB 200  
                  10  PLAY "e2."  
                  20  PLAY "???"  
                  30  PLAY "???"  
                  40  PLAY "???"  
                  50  PLAY "???"  
                  60  PLAY "?????"  
                  70  PLAY "???"  
                  80  PLAY "???"  
                  90  PLAY "?????????"  
                 100  PLAY "?????"  
                 110  PLAY "?????"  
                 120  PLAY "???"  
                 130  PLAY "???"  
                 140  PLAY "?????"  
                 150  PLAY "???"  
                 160  END  
                 200  A$="OH! MINAS GERAIS!  
                 210  B$="QUEM TE CONHECE"  
                 220  C$="NAO ESQUECE JAMAIS"  
                 230  SCREEN 0:KEY OFF  
                 240  PRINT A$,A$,B$,C$,A$  
                 250  RETURN
```

Qual a escala utilizada para compor esta melodia?

E 6.5- Antes da I Guerra Mundial, surgiu nos Estados Unidos, um estilo de piano denominado "RAGTIME" que chegou a dar nome a uma era. Seus maiores expoentes foram SCOTT JOPLIN e "JELLY-ROLL" MORTON.

Figura 6.23 - Scott Joplin  
(1868-1917)



Neste estilo, enquanto a mão direita tocava uma melodia, normalmente sincopada, a esquerda marcava o tempo, como se estivéssemos batendo um pano de pó (RAG, em inglês). A figura 6.24 mostra a partitura do famoso "THE ENTERTAINER" publicado por SCOTT JOPLIN em 1902 e que se tornou familiar à nossa geração pois foi o tema musical do "GOLPE DE MESTRE", estrelado por PAUL NEWMANN e ROBERT REDFORD.

Figura 6.24

The image displays two systems of musical notation, each consisting of a grand staff (treble and bass clefs) and a single staff below it. The notation includes various musical symbols such as notes, rests, and dynamic markings.

**System 1 (Left):**

- Staff 1 (Treble):** Starts with a *f* (forte) dynamic marking. It contains several measures of music, including a half note G4, a quarter note A4, and a half note B4. A *1 II accento* marking is present above the staff.
- Staff 2 (Bass):** Contains a half note G3, a quarter note A3, and a half note B3. A *f* (forte) dynamic marking is present below the staff.
- Staff 3 (Single):** Contains a half note G4, a quarter note A4, and a half note B4. A *f* (forte) dynamic marking is present below the staff.

**System 2 (Right):**

- Staff 1 (Treble):** Starts with a *f* (forte) dynamic marking. It contains several measures of music, including a half note G4, a quarter note A4, and a half note B4. A *1 II accento* marking is present above the staff.
- Staff 2 (Bass):** Contains a half note G3, a quarter note A3, and a half note B3. A *f* (forte) dynamic marking is present below the staff.
- Staff 3 (Single):** Contains a half note G4, a quarter note A4, and a half note B4. A *f* (forte) dynamic marking is present below the staff.

The first system of the musical score consists of two staves. The upper staff is in treble clef and contains a series of eighth and sixteenth notes, some beamed together, and a few rests. The lower staff is in bass clef and contains a series of eighth and sixteenth notes, some beamed together, and a few rests. There are two dynamic markings, *f* (forte), one in the middle of the upper staff and one in the middle of the lower staff. There are also two wedge-shaped markings, one in the middle of the upper staff and one in the middle of the lower staff, indicating a crescendo or decrescendo.

The second system of the musical score consists of two staves. The upper staff is in treble clef and contains a series of eighth and sixteenth notes, some beamed together, and a few rests. The lower staff is in bass clef and contains a series of eighth and sixteenth notes, some beamed together, and a few rests. There are two dynamic markings, *f* (forte), one in the middle of the upper staff and one in the middle of the lower staff. There are also two wedge-shaped markings, one in the middle of the upper staff and one in the middle of the lower staff, indicating a crescendo or decrescendo.

First system of a musical score. It consists of two staves. The upper staff contains a continuous melodic line with eighth and sixteenth notes, including some beamed pairs and a final half note. The lower staff contains a bass line with fewer notes, including a half note, a quarter note, and a half note, with some rests.

Second system of a musical score, featuring first and second endings. The system is divided into two main sections by a vertical line. The first section contains two staves with musical notation, including first and second ending brackets labeled '1.' and '2.'. The second section, to the right of the vertical line, also contains two staves with musical notation, including a first ending bracket labeled '1.'.



Estude então a partitura da figura 6.24 e transcreva-a para seu MSX, completando o programa da figura 6.25 .

Figura 6.25

```

80  SCREEN 0:KEY OFF
85  PRINT "THE ENTERTAINER"
86  PRINT "By Scott Joplin"
87  PRINT "1902"
90  PLAY "s0m3000T136      "
      "s0m3000T136      "
      "s0m3000T136      "
100 PLAY "o4d8d#8         "
      "r4                 "
      "r4                 "
105 FOR I=1 TO 2
110 PLAY "e8o5c4o4e8o5c4o4e8o5c8 "
      "o3L4ceg b-        "
      "r1L4              "
120 PLAY "c2r8c8d8d#8     "
      "faeg              "
      "ro4crc            "
130 PLAY "e8c8d8e4o4b8o5d4 "
      "go4co3gb          "
      "rerf              "
140 PLAY "c2r4o4d8d#8     "
      "o4?o3?o4?r4      "
      "errr              "
150 PLAY "e8o5c4o4e8o5c4o4e8o5c8 "
      "o3????-          "
      "rrrr              "
160 PLAY "????????????    "
      "????-            "
      "ro4?rr            "
170 PLAY "?#?????????????? "
      "??#??#           "
      "o3r?r?            "
180 PLAY "o5????o4???#?    "
      "??????           "
      "rrrr              "
190 PLAY "???????????????? "
      "?????-            "
      "rrrr              "
200 PLAY "c2r8c8d8d#8     "
      "????             "
      "o4r?r?            "

```

```

210 PLAY "?????????????????
      "?????????
      "???"
220 PLAY "c2r4c8d8
      "o4co3go4cr4
      "errr
230 PLAY "e8c8d8e4c8d8c8
      "cco3b-o4c
      "rere
240 PLAY "e8c8d8e4c8d8c8
      "?????????????????
      "???"
250 PLAY "?????????????????
      "?????????????
      "???"
255 IF I=2 THEN 270
260 PLAY "o5c2r4o4d8d#B
      "o4co3ged
      "rrrr
265 NEXT I
270 PLAY "o5c2.
      "o4co3gc
      "o3co2gc

```

E.6.6- Altere a linha 90 do programa anterior para o indicado na figura 6.26 .

Figura 6.26

```

90 PLAY "v10      T136
      "v9      T136
      "v9      T136

```

Ouç a música que o programa gera. O instrumento que o MSX simula agora mais parece um:

- a) piano
- b) violino
- c) órgão
- d) xilofone

E 6.7- Aproveitando a alteração feita no exercício anterior, agora podemos transcrever as indicações de intensidade (figura 6.27) que estão na partitura do THE ENTERTAINER.

Figura 6.27 - Notações de intensidade

***p*** = PIANO

***f*** = FORTE

**<** = CRESCENDO

**>** = DIMINUENDO

Isso ocorre porque, quando não usamos o envelope, passamos a ter o controle do volume de cada canal. Podemos, então, intercalar comandos de alteração de volume na 1ª voz de maneira a tornar a execução um pouco menos "mecânica". Para isso, no lugar do VXX que aparece nas linhas 120, 140, 160, 180, 200 e 260 da figura 6.28, você deverá digitar ou V 10 (piano) ou V 13 (crescendo ou diminuindo) ou V15 (forte).

Figura 6.28

```

120 PLAY "c2vXXr8c8d8d#8vXX      ",
      "faeg                          "
      "ro4crc                        "
140 PLAY "c2vXXr4o4d8d#8vXX      ",
      "o4co3go4cr4                  "
      "errr                          "
160 PLAY "c2vXXr4o4a8g8vXX      ",
      "faee-                         "
      "ro4crr                        "
180, PLAY "o5d2vXXr4o4d8d#8vXX   ",
      "go2gab                         "
      "rrrr                          "
200 PLAY "c2vXXr8c8d8d#8vXX      ",
      "faeg                          "
      "o4rcrc                        "
260 PLAY "o5c2vXXr4o4d8d#8vXX   ",
      "o4co3ged                      "
      "rrrr                          "

```

Faça as alterações obedecendo às indicações da partitura e ouça seu MSX tocar de maneira um pouco mais expressiva.


E 6.8- Na figura 6.29 temos um programa que toca uma das poucas músicas que poderíamos chamar de "universais".

Figura 6.29

```
5  PLAY "s0m6000o4T120"  
10 PLAY "g8.g16"  
20 PLAY "a4g4o5c4"  
30 PLAY "o4b2g8.g16"  
40 PLAY "a4g4o5d4"  
50 PLAY "c2e8.e16"  
60 PLAY "g4e4c4"  
70 PLAY "o4b4a4o5f8.f16"  
80 PLAY "e4c4d4"  
90 PLAY "c2."
```

Digite-o e rode-o. Reconheceu? Ótimo: agora complete a figura 6.30. gerando a partitura correspondentel.

Figura 6.30





## APÊNDICES

### APÊNDICE A

#### EXERCÍCIOS SUPLEMENTARES

Ao encerrar o estudo deste volume você já deve estar habilitado para transcrever para o MSX algumas partituras com um certo grau de complexidade.

A finalidade deste apêndice é fornecer alguns exercícios suplementares neste sentido.

Numa digitação longa, há sempre chance de ocorrer algum acidente: falta de força, sobrinho que tropeça no fio da tomada, ataques de insanidade mental, etc.

Lembre-se da lei de Murphy: "Se alguma coisa tem a mais remota chance de dar errado.... certamente dará errado"!

Porisso, mesmo que o programa não esteja terminado, é conveniente gravá-lo periodicamente de forma a ter sempre uma versão a mais atualizada possível.

E A.1- A partitura mostrada na figura A.1 corresponde ao famoso "TICO TICO NO FUBÁ" do saudoso Zequinha de Abreu. José Gomes de Abreu (1880-1935) nasceu em Santa Rita do Passa Quatro (SP) e notabilizou-se como pianista e compositor de choros, maxixes e valsas. A partitura em questão está adaptada para duas vozes em Clave de Sol sendo de fácil digitação (apesar de longa). Tome cuidado com os acidentes no começo da clave pois eles mudam no decorrer da música.

1.<sup>a</sup> VOZ

2.<sup>a</sup> VOZ

FIN

Musical score system 1 (top staff):  
 Treble clef, key signature of one sharp (F#), 2/4 time signature.  
 The staff contains several measures of music, including a triplet of eighth notes. A dynamic marking *f* is present.

Musical score system 2 (top staff):  
 Treble clef, key signature of one sharp (F#), 2/4 time signature.  
 The staff contains several measures of music, including a triplet of eighth notes. A dynamic marking *f* is present.

Musical score system 3 (top staff):  
 Treble clef, key signature of one sharp (F#), 2/4 time signature.  
 The staff contains several measures of music, including a triplet of eighth notes. A dynamic marking *f* is present.

Musical score system 4 (top staff):  
 Treble clef, key signature of one sharp (F#), 2/4 time signature.  
 The staff contains several measures of music, including a triplet of eighth notes. A dynamic marking *f* is present.

E A.2- Se você pedir a algum músico para assobiar um minueto, provavelmente você ouvirá o conhecidíssimo "MINUETO EM L4", mais conhecido, justamente como "Minueto de Boccherini". Sua partitura está na figura A.2 e para transcrevê-la para o MSX você precisará das 3 vozes. Luigi Boccherini (1743-1805) nasceu em Lucca, na Itália e é autor de várias sinfonias, sonatas, trios, quartetos e quintetos.



E A.3- Não podia faltar uma partitura ( figura A.3) do famoso JOHANN SEBASTIAN BACH ( 1685-1750). A música escolhida é o "JESUS ALEGRIA DOS HOMENS" rica em "terzinas" numa partitura para duas vozes. Ao transcrevê-la lembre-se que a semínima tem duração L4, as "pseudo-colcheias" da terzina têm duração L12.



1<sup>a</sup> VOZ

2<sup>a</sup> VOZ

*p*

*pp*

*rit.*

This musical score is written for two voices and piano accompaniment. The first system features two vocal staves, labeled '1<sup>a</sup> VOZ' and '2<sup>a</sup> VOZ', and a piano accompaniment staff. The piano part begins with a dynamic marking of *p* (piano). The second system continues the vocal and piano parts. The third system shows the vocal parts continuing, with the piano part featuring a *rit.* (ritardando) marking. The fourth system concludes the page, with the piano part marked *pp* (pianissimo) and a final *rit.* marking. The music is written in a common time signature and includes various musical notations such as notes, rests, and bar lines.

E.A.4- Para sair um pouco dos "clássicos" figura A.4. mostra a partitura de um "BLUE BOOGIE". Note que muitos compassos são iguais para a mão esquerda. Digite-os antes alterando o número de linha do BASIC, para poder aproveitar a digitação de uma linha em vários pontos.



## APÊNDICE B

### RESPOSTAS DOS EXERCÍCIOS

#### AULA 1

- E 1.1 - a) GREENLEAVES  
b) Anônimo  
c) Século XVIII

- E 1.3 - a) A decolagem de um avião a jato.  
b) O tempo de aquecimento do motor é menor.  
c) A velocidade de decolagem foi aumentada.

#### AULA 2

- E 2.1 - b) Altura

- E 2.2 - a) Intensidade

- E 2.3 - c) Duração

- E 2.4 - "ATIREI UM PAU NO GATO"

- E 2.5 - 30 PLAY "AGFFFGFEEE"

- E 2.6 - 40 PLAY "CCAAABAGGG"

- E 2.7 - 50 PLAY "EFGFEGFEDC"

#### AULA3

- E 3.1 - nota 2: PLAY"o3B"  
nota 3: PLAY"o4E"  
nota 4: PLAY"o4A"  
nota 5: PLAY"o5D"  
nota 6: PLAY"o5B"

- E 3.2 -

5 PLAY "s0m8000","s0m8000","s0m8000"  
10 PLAY "o5d8d8c4o4a4"  
20 PLAY "o5c8c8o4b4o5c4"  
30 PLAY "d2r4"  
40 PLAY "c8o4b8a4g4"  
50 PLAY "a8g8a4b4"  
60 PLAY "o5c2r4"  
70 PLAY "d8d8o4b-4g4"  
80 PLAY "o5d8d8c4o4a4"  
90 PLAY "a2r4"

## F 3.3 1

```

5  PLAY "s0m8000", "s0m8000"
10 PLAY "o5d8d8c4o4a4", "o4d4a2"
20 PLAY "o5c8c8o4b4o5c4", "a4a2"
30 PLAY "d2r4", "a4a2"
40 PLAY "c8o4b8a4g4", "g2r4"
50 PLAY "a8g8a4b4", "a4g2"
60 PLAY "o5c2r4", "a4g2"
70 PLAY "d8d8o4b-4g4", "a4b-2"
80 PLAY "o5d8d8c4o4a4", "a4a4g4"
90 PLAY "a2r4", "f8e8d2"

```

## F 3.4 -

```

5  PLAY "s0m8000","s0m8000","s0m8000"
10 PLAY "o5d8d8c4o4a4","o4d4a2","o3d2."
20 PLAY "o5c8c8o4b4o5c4","a4a2","f2."
30 PLAY "d2r4","a4a2","d2."
40 PLAY "c8o4b8a4g4","g2r4","c2r4"
50 PLAY "a8g8a4b4","a4g2","d2."
60 PLAY "o5c2r4","a4g2","f2."
70 PLAY "d8d8o4b-4g4","a4b-2","g2."
80 PLAY "o5d8d8c4o4a4","a4a4g4","d2."
90 PLAY "a2r4","f8e8d2","d2r4"

```

## E 3.5 -



E 3.6 - O comando PLAY presente na tela é executado.

#### AULA 4

E 4.2 - A linha 610. deixando-a assim:

```
610 FOR I=1 TO 5
```

E 4.3 - As linhas 500, 510, 600 e 690. deixando-as assim:

```
500 DIM Y(20)
510 FOR I=1 TO 20
600 FOR N=1 TO 20
690 LOCATE 3,20:PRINT"SUA NOTA=";X/2
```

E 4.4 - O valor é 120.

E 4.5 -

```
5 PLAY "s0m5000T120"
10 PLAY "o5e2a4.g16f16"
20 PLAY "o5e2a4.g16f16"
30 PLAY "e4a4a4o6c4"
40 PLAY "o5b12o6c12o5b12a12g12a12b4g4"
50 PLAY "o5e2a4.g16f16"
60 PLAY "e2o4a4.g16f16"
70 PLAY "e2e4.d8"
80 PLAY "e2.g4"
90 PLAY "t140a4b4o5c4d4"
100 PLAY "c4c4c4o4o4b4"
110 PLAY "a4a4a4g4"
120 PLAY "a2.a4"
```

E 4.6 -

```
5 PLAY "s0m5000T120","v10T120","v12T120"
10 PLAY "o5e2a4.g16f16","o3g2e2","o3c2o2a2"
20 PLAY "o5e2a4.g16f16","o3g2e2","o3c2o2a2"
30 PLAY "e4a4a4o6c4","o3g2e2","o3c2o2a2"
40 PLAY "o5b12o6c12o5b12a12g12a12b4g4","o2b12o4c12o2b12a12g12a12b4g4","r1"
50 PLAY "o5e2a4.g16f16","o3g2e2","o3c2o2a2"
```

```

60 PLAY "e2o4a4.g16f16","o3g2e2","o3c2o2
a2"
70 PLAY "e2e4.d8","o3g2g2","o3c2o2b2"
80 PLAY "e2.g4","o3g2.r4","o3c2.r4"
90 PLAY "T140a4b4o5c4d4","T140r1","T140r
1"
100 PLAY "c4c4c4o4o4b4","o3c8e8g8e8o3c8e
8g8e8","r1"
110 PLAY "a4a4a4g4","o3c8e8g8e8o3c8e8g8e
8","r1"
120 PLAY "a2.a4","o3d8f8a8f8o3d8f8a8f8",
"r1"

```

E 4.7 -

```

5    PLAY "v11T100      "
      "s0m4000T100      "
      "s0m4000T100      "
6    FOR I=1 TO 2
8    PLAY "o4e8e8      "
      "o3r4      "
      "o3r4      "
10   PLAY "g4.e8e4      "
      "c4g4g4      "
      "r4e4e4      "
20   PLAY "g2e8g8      "
      "c4g4g4      "
      "r4e4e4      "
30   PLAY "o5c4o4b4.a8      "
      "c4g4g4      "
      "r4e4e4      "
40   PLAY "a4g4d8e8      "
      "o2b4o3g4g4      "
      "r4f4f4      "
50   PLAY "f4d4d8e8      "
      "o2b4o3g4g4      "
      "r4f4f4      "
60   PLAY "f2d8f8      "
      "o2b4o3g4g4      "
      "r4f4f4      "
70   PLAY "b8a8g4b4      "
      "o3d4o3g4g4      "
      "r4f4f4      "
80   PLAY "o5c2o4c8c8      "
      "c4g4r4      "
      "r4e4r4      "
90   PLAY "o5c2o4a8f8      "
      "c4a4a4      "
      "r4f4f4      "

```



## AULA 5

## E 5.1 -

```

5   PLAY "s0m9000" "s0m9000" "s0m9000"
6   PLAY "T104" "T104" "T104"
7   SCREEN 0:PRINT"EM MODO DORICO"
8   PRINT "BELA BARTOK"
10  PLAY "f4o5d4c4o4b4a2",
      "o4d1d4g4"
20  PLAY "a4b4o5c4o4g4a2",
      "d1d4c4"
30  PLAY "a4g4a4b4o5c4d4",
      "d2.d4e4f4"
40  PLAY "o4b1bgo5c4",
      "g4d4e4f4g4e4"
50  PLAY "o4a2.o5d4o4a4b4",
      "g4c4d4e4f2"
60  PLAY "g4a4b4o5c4o4a4r4",
      "f2d4e4c4f4"
70  PLAY "r4o5c4o4b2g2",
      "d4c4e2g2"
80  PLAY "r2a2o5c4r4",
      "d2.c4e4g4"
90  PLAY "r4d4o4a2o5c4r4",
      "d2.c4e4g4"
100 PLAY "r4o4g4a4o5d4c4r4",
      "d2.c4e4g4"
110 PLAY "r4o4g4a4o5d4c4o4b4",
      "d2.c4e4g4"
120 PLAY "a4b4o5d4o4g4a4o5d4",
      "g4f4e4g2f4"
130 PLAY "c4o4b4a2g2",
      "e4c4f2e2"
140 PLAY "a1.",
      "f#1."
      "d1."

```

## E 5.2 -

```

4   PLAY "s0m20000" "s0m20000"
5   PLAY "T80" "T80"
6   SCREEN 0:KEY OFF:WIDTH 39
7   PRINT"BELA BARTOK:"
8   PRINT"EM MODO FRIGIO"
9   PRINT"CALMO"
10  PLAY "r4o5c4o4b2",
      "o3e2.f4"

```



20	PLAY "b4o5d4o4b2	" "
	"e2.g4	" "
30	PLAY "b4o5e4o4b2	" "
	"e2.a4	" "
40	PLAY "b2o5d4e4f4	" "
	"e1	" "
50	PLAY "e4o4b4o5e2	" "
	"e4f4g4a4	" "
60	PLAY "d2o4b4o5d4	" "
	"b1	" "
70	PLAY "c1	" "
	"b4a4g4a4	" "
80	PLAY "c1	" "
	"b4a4g4f4	" "
90	PLAY "r4o4b4o5d4e4	" "
	"e2b2	" "
100	PLAY "f2c2	" "
	"b4a4b2	" "
110	PLAY "d2r4o4b4	" "
	"f2g2	" "
120	PLAY "o5c4d4e2	" "
	"e2.b4	" "
130	PLAY "o4b2o5c2	" "
	"a4g4a2	" "
140	PLAY "r4d4o4b2	" "
	"f2g2	" "
150	PLAY "o5e4c2d4	" "
	"e2f2	" "
160	PLAY "o4b2o5e4c4	" "
	"g2f2	" "
170	PLAY "c4d4o4b2	" "
	"g2a2	" "
180	PLAY "o5c4o4b4r4o5c4	" "
	"g2a2	" "
190	PLAY "o4b4o5d4e4f4	" "
	"b1	" "
200	PLAY "e4o4b4o5e2	" "
	"b4a4g4a4	" "
210	PLAY "d2o4b4o5d4	" "
	"b1	" "
220	PLAY "c4o4b4o5c2	" "
	"b4a4g4a4	" "
230	PLAY "o4b1	" "
	"f2b2	" "
240	PLAY "b1	" "
	"g2a2	" "
250	PLAY "b1	" "
	"e1	" "
260	PLAY "b1	" "
	"e1	" "

E 5.3 -

```

5  PLAY "s0m5000L4T200o4"
10 PLAY "cdefrffr"
20 PLAY "cdcdrddr"
30 PLAY "cgfereer"
40 PLAY "cdefrffr"
50 PLAY "o3gabo4crccr"
60 PLAY "o3gagaraar"
70 PLAY "go4dco3brbbr"
80 PLAY "o3gabo4crccr"

```

E 5.4 -

```

5  PLAY "s0m5000L4T200o4"
10 PLAY "cdefrffr"
20 PLAY "cdcdrddr"
30 PLAY "cgfereer"
40 PLAY "cdefrffr"
50 PLAY "o3gabo4crccr"
60 PLAY "o3gagaraar"
70 PLAY "go4dco3brbbr"
80 PLAY "o3gabo4crccr"
90 PLAY "fgabrbbbr"
100 PLAY "fgfgrggr"
110 PLAY "fo5co4baraar"
120 PLAY "fgabrbbbr"

```

E 5.5 -

```

5  PLAY "s0m5000L4T200o4"

10 PLAY "cdefrffr"
20 PLAY "cdcdrddr"
30 PLAY "cgfereer"
40 PLAY "cdefrffr"
50 PLAY "o3gabo4crccr"
60 PLAY "o3gagaraar"
70 PLAY "go4dco3brbbr"
80 PLAY "o3gabo4crccr"
90 PLAY "fgab-rb-b-r"
100 PLAY "fgfgrggr"
110 PLAY "fo5co4b-araar"
120 PLAY "fgab-rb-b-r"

```

## AULA 6

E 6.1 - No Expert versão 1.1 :

B N K , - ; \*

E 6.2 - No Expert versão 1.1 :

G H M K L Ç / \*

E 6.3 -



E 6.4 -

```

5   PLAY "s0m9000T13Bo4"
6   GOSUB 200
10  PLAY "e2."
20  PLAY "d4c4d4"
30  PLAY "c2."
40  PLAY "c2r4"
50  PLAY "o5c2."
60  PLAY "o4a4f4a4"
70  PLAY "g2."
80  PLAY "g2r4"
90  PLAY "o5c4o4a4o5c4"
100 PLAY "o4b4g4b4"
110 PLAY "a4f4a4"
120 PLAY "g2."
130 PLAY "e2."
140 PLAY "d4c4d4"
150 PLAY "c2."
160 END
200 A$="OH! MINAS GERAIS!
210 B$="QUEM TE CONHECE"
220 C$="NAO ESQUECE JAMAIS"
230 SCREEN 0:KEY OFF
240 PRINT A$,A$,B$,C$,A$
250 RETURN
    
```

E 6.5 -

```

80  SCREEN 0:KEY OFF
85  PRINT "THE ENTERTAINER"
86  PRINT "By Scott Joplin"
87  PRINT "1902"
90  PLAY "s0m3000T136      " "
      "s0m3000T136      " "
      "s0m3000T136      " "
100 PLAY "o4d8d#8        " "
      "r4                " "
      "r4                " "
105 FOR I=1 TO 2
110 PLAY "e8o5c4o4e8o5c4o4e8o5c8 " "
      "o3L4ceg b-        " "
      "r1L4              " "
120 PLAY "c2rBc8d8d#8    " "
      "faeg              " "
      "ro4crc            " "
130 PLAY "e8c8d8e4o4b8o5d4 " "
      "go4co3gb          " "
      "rerf              " "
140 PLAY "c2r4o4d8d#8    " "
      "o4co3go4cr4       " "
      "errr              " "
150 PLAY "e8o5c4o4e8o5c4o4e8o5c8 " "
      "o3ceg b-          " "
      "rrrr              " "
160 PLAY "c2r4o4aBg8     " "
      "faee-             " "
      "ro4crr            " "
170 PLAY "f#B8a8o5c8e4d8c8o4a8 " "
      "df#d f#           " "
      "o3rara             " "
180 PLAY "o5d2r4o4d8d#8  " "
      "go2gab             " "
      "rrrr              " "
190 PLAY "e8o5c4o4e8o5c4o4e8o5c8 " "
      "o3ceg b-          " "
      "rrrr              " "
200 PLAY "c2r8c8d8d#8    " "
      "faeg              " "
      "o4rcrc            " "
210 PLAY "e8c8d8e4o4b8o5d4 " "
      "go4co3gb          " "
      "rerf              " "
220 PLAY "c2r4c8d8       " "
      "o4co3go4cr4       " "
      "errr              " "

```


230	PLAY	"e8c8d8e4c8d8c8	"
		"cco3b-o4c	"
		"rere	"
240	PLAY	"e8c8d8e4c8d8c8	"
		"o3ao4co3b-o4c	"
		"rfrf	"
250	PLAY	"e8c8d8e4o4b8o5d4	"
		"o3go4co3gb	"
		"rerr	"
255	IF	I=2 THEN 270	
260	PLAY	"o5c2r4o4d8d#8	"
		"o4co3ged	"
		"rrrr	"
265	NEXT	I	
270	PLAY	"o5c2.	"
		"o4co3gc	"
		"o3co2gc	"

E 6.6 - c) órgão

E 6.7 -

120	PLAY	"c2v13r8c8d8d#8v15	"
		"faeg	"
		"ro4crc	"
140	PLAY	"c2v13r4o4d8d#8v10	"
		"o4co3go4cr4	"
		"errr	"
160	PLAY	"c2v13r4o4a8g8v15	"
		"faee-	"
		"ro4crr	"
180	PLAY	"o5d2v13r4o4d8d#8v10	"
		"go2gab	"
		"rrrr	"
200	PLAY	"c2v13r8c8d8d#8v15	"
		"faeg	"
		"o4rcrc	"
260	PLAY	"o5c2v13r4o4d8d#8v10	"
		"o4co3ged	"
		"rrrr	"

E 6.8 -



# APENDICE A

E A1 -

5 I=0

10 SCREEN1:BASE(9)=0:SCREEN1:KEYOFF

20 INTERVAL ON

30 ON INTERVAL=10 GOSUB 1000

40 PRINT "TICO TICO NO FUBA"

50 PRINT "ZEQUINHA DE ABREU"

100 PLAY "s0m2000T80","v11T80"

110 PLAY "o4a16g#16a16","o4f16e16f16"

120 PLAY "b-16a16o5d8.o4a16g#16a16","g16f16a16f16d16f16e16f16"

130 PLAY "b-16a16o5c#8.o4a16g#16a16","g16e16a16g16e8r8"

140 PLAY "b-16a16o5g16e16c#16o4a16g16f#16","e4g8e8"

150 PLAY "f4r16o5d16c#16c16","d16e16f16a16o5d8r8"

160 PLAY "o4b-16o5d16g8.f16e16d16","o4g16a16b-16a16g8r8"

170 PLAY "o4a16o5d16f8.f16e16d#16","f16g16a16g16f8r8"

180 PLAY "e16o4e16g#16b16o5d16f16e16d16","g#4b4"

190 PLAY "c#16a8e16a16o4a16g#16a16","a16o5c#8o4a16o5c#16o4f16e16f16"

200 PLAY "b-16a16o5d8.o4a16g#16a16","g16  
 f16a16f16d16f16e16f16"  
 210 PLAY "b-16a16o5c#8.o4a16g#16a16","g1  
 6e16a16g16e8r8"  
 220 PLAY "b-16a16o5g16e16c#16o4a16g16f#1  
 6","e4g8e8"  
 230 PLAY "f4r16o5d16c#16c16","d16e16f16a  
 16o5d8r8"  
 240 PLAY "o4b-16o5d16g8.f16e16d16","o4g1  
 6a16b-16a16g8r8"  
 250 PLAY "o4a16o5d16f8.f16e16d#16","f16g  
 16a16g16f8a8"  
 260 PLAY "e16o4a16o5c#16e16a16g16f16e16"  
 ,"g4o5c#16o4b-16a16g16"  
 265 IF I=1 THEN GOTO 610  
 270 PLAY "d4r16o4d16f#16a16","f16g16a16f  
 16d8r8"  
 280 PLAY "o5d16o4d16f#16o5c#8o4d16f#16b1  
 6","o4e8.a8.e8"  
 290 PLAY "b16o4d16f#16a8d16f#16b16","d8.  
 f#16f#4"  
 300 PLAY "b16o4d16f#16a8d16f#16b16","d8.  
 f#8.d8"  
 310 PLAY "b16o4e16g16a16r16g16a16b16","e  
 8.g16r16e16f#16g16"  
 320 PLAY "o5d16o4g16a16o5c#8o4g16a16b16"  
 ,"e8.a8.e8"  
 330 PLAY "b16o4e16g16a8g16a16b16","g8.e8  
 e16f#16g16"  
 340 PLAY "o5d16o4g16a16o5c#8o4g16a16b16"  
 ,"e8.a8.e8"  
 350 PLAY "b16d16f#16a16r16d16f#16a16","d  
 8.f#16r4"  
 360 PLAY "o5d16o4d16f#16o5c#8o4d16f#16b1  
 6","f#8.a8.f#8"  
 370 PLAY "b16o4d16f#16a8d16f#16b16","d8.  
 f#8.d8"  
 380 PLAY "b16a16f#16d#16b16a16f#16d#16",  
 "d#4f#4"  
 390 PLAY "e16d#16e16f#16g8r8","g16f#16g1  
 6a16b8r8"

400 PLAY "g1Bf#16g16a16b16o5c#16d16e16",  
 "c16d#16e16f#16g16a16b16l5c#16"  
 410 PLAY "f#16g16g#16a16b16a16g16f#16",  
 "d16e-16e16f#16g16f#16e16d16"  
 420 PLAY "e16d16c#16o4b16a16g16f#16e16",  
 "c#16o4b16a16g16f#16e16d16c#16"  
 430 PLAY "d4r16e16d16d-16", "d8a8o5d8r8"  
 440 PLAY "c16f16a16o5c16o4f16a16o5c16c#1  
 6", "f4a4"  
 450 PLAY "d8o4b-8.e16d16d-16", "g16a16b-1  
 6g16e8r8"  
 460 PLAY "c16e16g16b-16e16g16b-16o5c16",  
 "e4g4"  
 470 PLAY "d8o4a8.o5f16f16f16", "f16g16a16  
 f16o5c16d16d16d16"  
 480 PLAY "f16e16e16e16e16d16d16d16", "d16  
 c16c16c16c16o4b-16b-16b-16"  
 490 PLAY "d4d16e16e16e16", "b-4b-16o5c16c  
 16c16"  
 500 PLAY "e16d16d16d16d16c16c16c16", "c16  
 o4b-16b-16b-16b-16a16a16a16"  
 510 PLAY "c4r16o4a16f16d16", "a16f16a16o5  
 d16c8r8"  
 520 PLAY "c16f16a16o5c16o4f16a16o5c16c#1  
 6", "f4a4"  
 530 PLAY "o5d8o4b-8.e16d16d-16", "g16a16b  
 -16g16e8r8"  
 540 PLAY "c16e16g16b-16e16g16b-16o5c16",  
 "e4g4"  
 550 PLAY "d8o4a8.o5f16e16e-16", "f16g16a1  
 6f16c16f16g16a16"  
 560 PLAY "d16c#16d16e16g16f16e16f16", "b-  
 2"  
 570 PLAY "a16c16f16o4a16o5c16o4b16b-16a1  
 6", "o5c16o4a16o5c16o4f16a16g#16g16f16"  
 580 PLAY "g16o5f16e16d16c16o4b-16a16g16",  
 "e8g16b-16o5e16d16c16o4b-16"  
 590 PLAY "f4r16a16g#16a16", "a4r16f16e16f  
 16"  
 600 I=1:GOTO 120



```

610 PLAY "o5d4r16","o4f16g16a16f16d8"
620 IF PLAY(1) THEN 620
630 SCREEN 1:PRINT"fim":END
1000 PUT SPRITE INT(RND(1)*30+1),(RND(1)
*255,RND(1)*192),INT(RND(1)*15+1),14
1010 RETURN

```

E A2 -

```

50 SCREEN0,,0:KEY OFF
60 PRINT "MINUETO EM LA"
70 PRINT "L.BOCCHERINI"
80 PRINT "(1743-1805)"
90 PLAY "s0m5000T72","s0m5000T72","s0m5
000T72"
100 PLAY "q5a16g#16a16b16","r4","r4"
110 PLAY "a8o4a4o5c#4e8","o4c#8e8r8e8c#8
r8","o3a8r8e8r8r8a8"
120 PLAY "e8d8d4d16c#16d16e16","o4d8e8r8
e8d8e8","o3g#8r8e8r8g#8r8"
130 PLAY "d8o4e4b4o5d8","o4d8r8r8e8o3b8o
4e8","o3g#8r8e8r8g#8r8"
140 PLAY "d8c#8c#4a8.f#16","o3a8o4e8c#8e
8c#8o3a8","o4r4r4a4"
150 PLAY "e8d#8d#8d#8a8.f#16","o3f#8b8a8
b8o4d#8o3b8","o4a2a4"
160 PLAY "e8d#8d#8d#8a8.f#16","o3f#8b8a8
b8o4d#8o3b8","o4a2b4"
170 PLAY "g#8e8c#8a8g#32f#32g#32f#32g#32
f#32e32f#32","o3b8o4e8c#8e8d#4","o4b4o3a
8r8o4a4"
180 PLAY "e4.r8","o4g#8o3b8e8r8","e8r8r8
r8"

```

```

B0 SCREEN0:KEY OFF
B5 PRINT "JESUS, ALEGRIA DO HOMENS"
90 PRINT "JOHANN SEBASTIAN BACH"
95 PRINT "(1685-1750)"
100 PLAY "s0m6000TB0" " "
      "v12TB0","v12TB0"
110 PLAY "o4r8g8a8L12bo5dccc" " "
      "o4r8e8f#8L12gbaao5co4b" " "
120 PLAY "dgf#gdo4bgab" " "
      "bo5ed#eo4bgef#g" " "
130 PLAY "o5cdedco4babg" " "
      "abo5co4bagf#ge" " "
140 PLAY "f#gadf#ao5co4ba" " "
      "def#dr12f#agf#H" " "
150 PLAY "bgabo5dccc" " "
      "gef#gbaao5co4b" " "
160 PLAY "dgf#gdo4bgab" " "
      "bo5ed#eo4bgef#g" " "
170 PLAY "eo5dco4bagdgf#H" " "
      "c4c#4d4" " "
180 PLAY "gbo5dgdo4bgbo5d" " "
      "L12egbo5eo4bgegb" " "
190 PLAY "fdo4bgbo5deco4a" " "
      "o5do4bgfgbo5co4ae" " "
200 PLAY "f#ao5cdo4bgegb" " "
      "d2e6g12" " "
210 PLAY "o5co4af#df#ao5co4ba" " "
      "f#4L12f#df#agf#H" " "
220 PLAY "bgabo5dccc" " "
      "gef#gbaao5co4b" " "
230 PLAY "dgf#gdo4bgab" " "
      "bo5ed#eo4bgef#g" " "
240 PLAY "o5cdedco4babg" " "
      "abo5co4bagf#ge" " "
250 PLAY "f#gadf#ao5co4ba" " "
      "def#dr12f#agf#H" " "
260 PLAY "bgabo5dccc" " "
      "gef#gbaao5co4b" " "

```

```

270 PLAY "dgf#gdo4bgab      "
      "bo5ed#eo4bgef#g      "
280 PLAY "eo5dco4bagdgf#    "
      "c4c#4L12ded          "
290 PLAY "g2r4              "
      "b2r4","g2r4"
300 IF PLAY(0) THEN 300
310 LOCATE 11,18:PRINT"FIM"

```

E A4 -

```

80 SCREEN 0:KEY OFF:PRINT"BLUE BOOGIE"
90 PRINT "JOHN BRIMHALL"
100 PLAY "v13T180","s0m5000T180","s0m500
0T180"
110 PLAY "o2c4o3c8.o2e16g4e8.g16","r1","
r1"
120 PLAY "o2c4o3c8.o2e16g4e8.g16","r1","
r1"
130 PLAY "o2c4o3c8.o2e16g4e8.g16","o5c1"
,"o4e1"
140 PLAY "o2c4o3c8.o2e16g4e8.g16","o5c4c
4c4d4","o4e4e4e4f4"
150 PLAY "o2c4o3c8.o2e16g4e8.g16","o5e2.
e8.f16","o4g2.g8.a16"
160 PLAY "o2c4o3c8.o2e16g4e8.g16","o5g2.
a4","o4b-2.o5c4"
170 PLAY "o2f4o3f8.o2a16o3c4o2a8.o3c16",
"o5f1","o4a1"
180 PLAY "o2f4o3f8.o2a16o3c4o2a8.o3c16",
"o5f4f4r2","o4a4a4o5c4d4"
190 PLAY "o2c4o3c8.o2e16g4e8.g16","o5e1"
,"o4g1"
200 PLAY "o2c4o3c8.o2e16g4e8.g16","o5e4r
4r4a4","o4g4o5c4e4c4"
210 PLAY "o2g4o3g8.o2b16o3d4o2b8.o3d16",
"o5g2.g8.g-16","o4b2.b8.b-16"
220 PLAY "o2f4o3f8.o2a16o3c4o2a8.o3c16",
"o5f2.r4","o4a2.o5c8.d16"
230 PLAY "s0m5000o2c4o3c8.o2e16g4e8.g16v
14","o5e2.r4","o4g2.r4"

```

```

240 PLAY "s0m5000o2g8.g16f8.f16e8.e16d8.
d16v13","r1","r1"
250 PLAY "o2c4o3c8.o2e16g4e8.g16","o4c1"
,"o3e1"
260 PLAY "o2c4o3c8.o2e16g4e8.g16","o4c4c
4c4d4","o3e4e4e4f4"
270 PLAY "o2c4o3c8.o2e16g4e8.g16","o4e2.
e8.f16","o3g2.g8.a16"
280 PLAY "o2c4o3c8.o2e16g4e8.g16","o4g2.
a4","o3b-2.o4c4"
290 PLAY "o2f4o3f8.o2a16o3c4o2a8.o3c16",
"o4f1","o3a1"
300 PLAY "o2f4o3f8.o2a16o3c4o2a8.o3c16",
"o4f4f4r2","o3a4a4o4c4d4"
310 PLAY "o2c4o3c8.o2e16g4e8.g16","o4e1"
,"o3g1"
320 PLAY "o2c4o3c8.o2e16g4e8.g16","o4e4r
4r4a4","o3g4o4c4e4c4"
330 PLAY "o2g4o3g8.o2b16o3d4o2b8.o3d16",
"o4g2.g8.g-16","o3b2.b8.b-16"
340 PLAY "o2f4o3f8.o2a16o3c4o2a8.o3c16",
"o4f2.r4","o3a2.o4c8.d16"
350 PLAY "o2c4o3c8.o2e16g4e8.g16","o4e1"
,"o3g1"
360 PLAY "o2c4o3c8.o2e16g4e8.g16","o4e4c
4e4a4","o3g4r4r4o4c4"
370 PLAY "o2g4o3g8.o2b16o3d4o2b8.o3d16",
"o4g2.g8.g-16","o3b2.b8.b-16"
380 PLAY "o2f4o3f8.o2a16o3c4o2f4","o4f2.
f4","o3a2.a-4
"
390 PLAY "o2c4o3c8.o2e16g4e8.g16","o4e1"
,"o3g1"
400 PLAY "o2c4c4c4r4","o4e4r4r2","o3g4r4
r2"
410 IF PLAY(0) THEN 410
420 LOCATE 10,10:PRINT"FIM"

```

## APÊNDICE C

### 1) O COMANDO PLAY

#### FUNÇÃO

O comando PLAY do MSX permite tocar sequências de notas ou acordes musicais compostos de uma, duas ou três notas simultâneas, com andamento, duração, timbre ou volume programáveis.

#### SINTAXE:

A Sintaxe do comando PLAY é

PLAY A\$, B\$, C\$

onde A\$, B\$ e C\$ são variáveis "strings" (alfanuméricas) que contém uma sequência de caracteres correspondentes aos subcomandos especificados adiante. A\$ é executada no primeiro canal de voz, B\$ no segundo e C\$ no terceiro.

A sequência de sub-comandos pode ser definida logo após o comando PLAY, sendo digitada entre aspas, ou pode ser atribuída a uma variável string num trecho anterior do programa BASIC.

#### Exemplos:

1) 10 PLAY "C D E F G A B "

2) 10 A\$ = " C D E F G A B "  
20 PLAY A\$

A variável string que determina a sequência de sub-comandos a serem executados pelo PLAY pode ser obtida pela "concatenação" de strings menores.

Devemos ter sempre o cuidado de transformar variáveis numéricas em variáveis alfanuméricas usando a função STR\$.

Digamos, por exemplo, que se queira executar a seguinte sequência de sub-comandos:

PLAY " T 150 V 12 AB "

onde os números 150 e 12 são obtidos de variáveis numéricas definidas em outros pontos do programa. Neste caso montaremos a variável string correspondente, conforme indica o exemplo a seguir.

```

10 X = 150
20 Y = 12
30 A$= "T" + STR$(X) + "V" + STR$(Y) + "AB"
40 PLAY A$

```

Uma outra maneira de se montar a string de sub-comandos a partir de variáveis anteriormente definidos no programa é a de se inserir estas variáveis com delimitadores especiais.

Quando a variável é alfanumérica ela deve ser precedida por um "X" e seguida por ";". Quando é numérica deve ser precedida por um "=" e seguida por ";".

No exemplo anterior, podemos montar a string como segue:

```

10 A=150
20 B=12
30 C$="AB"
40 PLAY"T=A;V=B;XC$;"


```

## OS SUB-COMANDOS DO PLAY

Em todos os sub-comandos, que veremos a seguir podem ser utilizados, indiferentemente letras maiúsculas ou minúsculas,

### A - G

As letras de A à G determinam a nota a ser usada conforme as cifras da notação anglo-saxônica. Colocando um sinal # ou + logo após a letra, a altura será elevada de meio tom (sustenido). Colocando um sinal - (menos) sua altura abaixará meio tom (bemo).



### Ox

Este sub-comando especifica a oitava em que se situam as notas a serem executadas. O valor de X pode estar entre 1 e 8.

Ao ser ligado o micro, o valor inicial que ele assume ("default") é 04:



O programa a seguir executa a escala musical de Dó a Si desde a oitava 1 (mais grave) até a 8 (mais aguda):

```

10 A$ = " C D E F G A B "
20 FOR I = 1 TO 8
30 B$ = " 0 " + STR$ (I)
40 PLAY B$ + A$
50 NEXT I

```

## Lx

Este sub-comando determina a duração de uma nota a ser tocada. O valor de x pode estar entre 1 (duração mais longa) e 64 (mais curta).

A correspondência entre os valores de Lx e as figuras musicais é dada na figura a seguir:



A nota de maior duração é a semibreve ( L1 ). Os valores de x indicam o denominador da fração de semibreve que a nota dura. Por exemplo, a semínima tem duração L4. Isto significa que sua duração é 1/4 da semibreve.

A duração da nota pode, também, ser especificada após a cifra. Por exemplo: L 8 A equivale a A 8. Se não houver especificação de duração após a cifra vale a especificada no último L.

Os valores de x não precisam ser, obrigatoriamente, potências inteiras de 2 (1,2,4,8,... etc).

Isso permite, por exemplo, especificar a duração de "tercinas".

Imagine este compasso.



Como o compasso é 3/4 nele cabem 3 semínimas ( 3 notas de duração L4 ). Como o segundo tempo têm 3 notas, cada uma deverá, durar 1/3 de 1/4, ou seja 1/2. Sua duração, portanto, será L 12. O compasso acima será executado por,

ou      PLAY   " o4 g 4 a 12 g 12 f 12 b 4 "

ou      PLAY   " o4 L 12 g 4 a g f b 4 "

ou      PLAY   " o4 L   4 g a 12 g 12 f 12 b "

Colocando um ponto (.) após a especificação de duração esta passa a valer 1 + 1/2 vezes a duração anterior. Cada ponto adicional equivale a um acréscimo da metade do ponto anterior.

Exemplos,

A1.    : duração de  $1 + 1/2 = 3/2$  de A 1  
A1.. : duração de  $1 + 1/2 + 1/4 = 7/4$  de A 1

Ao ligarmos o computador o valor inicial ("default") é L 4.

Tx








Este sub-comando especifica o andamento da música (adagio, allegro, presto, etc) correspondendo à marcação de um "metrônomo interno" do MSX. O valor de x pode variar de 32 (lento) a 255 (rapidíssimo).

Ao ligarmos o computador o valor inicial ("default") é T 120, que corresponde a 120 semínimas por minuto.



## Rx

Especifica a duração das pausas (silêncios). O valor de x, como no sub-comando Lx, especifica a duração da pausa, indo de 1 a 64 conforme a correspondência a seguir:

						
R1	R2	R4	R8	R16	R32	R64

Podem ser usados valores inteiros que não precisam, necessariamente, potências inteiras de 2 (1,2,4,8,16...etc).

Se nada for especificado após o R, o MSX assume o valor R 4.

## Vx


Determina o volume (intensidade) de cada canal de som do MSX. O valor de x varia de 0 (silêncio) a 15 (volume máximo). Quando usamos esse sub-comando num dos 3 canais, o MSX passa a emitir neste canal som de órgão, com maior ou menor intensidade, conforme o valor de Vx utilizado.

## Nx

Este sub-comando representa uma alternativa à notação de cifras para especificar uma determinada nota. O valor de x pode variar de 0 a 96. Quando x=0 o sub-comando N0 corresponde a uma pausa.

De N 1 a N 96 temos todos os sons possíveis de serem obtidos pelo comando PLAY.

Um acréscimo de uma unidade ao x implica na subida de um semitom na altura de nota. Como existem 12 semi-tons numa oitava, somando-se ou subtraindo-se 12 ao x de Nx obter-se-á a mesma nota uma oitava acima ou uma oitava abaixo. Na oitava 4 (valor default do MSX) a correspondência entre os valores de x no Nx e as notas musicais é dada na figura a seguir:



Observe que a nota mais grave do PLAY (N 1) corresponde a um dó $\sharp$  (  $1 + 12 + 12 + 12 = 37$  ) e a mais aguda ( N 96 ) é um Si $\sharp$  ( Dó ) pois  $96 - 12 - 12 - 12 - 12 = 48$ .

Obs.: Quando colocamos o MSX numa certa oitava, ao comandarmos B (SI $\sharp$  SUSTENIDO) ele não toca o Dó da oitava seguinte mas sim o Dó do começo da oitava em que ele se encontra. Desta forma, a única maneira de se obter o Dó da oitava 9 ( que não existe no MSX ) é comandarmos N 96.

Para saber a que oitava pertence a nota Nx, basta fazer a seguinte operação:

$$\text{OITAVA} = \text{INT} ( x / 12 ) + 1$$

## Sx e My

O sub-comando Sx permite determinar a forma do "envelope" do tom emitido, ou seja, como o volume varia com o tempo durante a emissão da nota. Quando esse sub-comando é usado num canal, desativa-se o efeito do Vx, e vice-versa.

O sub-comando My, que sempre está associado ao Sx, determina o período de ação do envelope.

Exemplificando:

O sub-comando S 13 faz com que o volume do canal, em função do tempo, varie conforme o diagrama a seguir:



Aumentando o valor de y no sub-comando My, esta variação ocorre ao longo de um período maior,



Note que, neste exemplo, apesar do volume final do envelope ser V 15, nem sempre a nota emitida atinge esta intensidade final pois sua duração pode não alcançar o final do envelope:



Porisso a escolha do Sx e My deve ser extremamente criteriosa, caso contrário algumas notas praticamente "desaparecem" durante a execução.

O valor de y no My (período do envelope) pode variar de 0 a 65535 enquanto que o x do Sx (forma do envelope) pode variar de 0 a 15.

As formas dos envelopes que podem ser obtidas com o sub-comando Sx e são dadas na figura a seguir:

VALOR ATRIBUÍDO AO REGISTRO 13 (valores atribuídos a Sn da função PLAY)	FORMA DO ENVELOPE
0,1,2,3 ou 9	
4,5,6,7 ou 15	
8	
10	
11	
12	
13	
14	

## 2) A FUNÇÃO PLAY

Quando uma sequência de sub-comandos é utilizada pelo comando PLAY, o MSX não a envia diretamente ao PSG (gerador programável de sons) mas sim a uma região de memória denominada BUFFER do PSG. Feito isso, o microprocessador se considera liberado e passa a executar o resto do programa BASIC.

Portanto o PSG vai lendo e esvaziando o BUFFER, tocando a música enquanto o micro já está realizando outra tarefa.

Isso pode causar efeitos indesejáveis de falta de sincronização. Para evitar isso, podemos utilizar a função

PLAY ( C )

onde C é o número do canal que queremos testar. Se o canal estiver ativo, a função assume o valor -1. se estiver mudo (BUFFER vazio) o valor da função será 0.

Se quisermos "congelar" o programa até que, por exemplo, o canal 2 tenha terminado sua execução, devemos usar uma linha de BASIC do tipo

```
575 IF PLAY (2) = -1 THEN GOTO 575
```

Neste caso a execução do programa em BASIC ficará parada na linha 575 enquanto o BUFFER do canal 2 contenha códigos a serem enviados para o PSG. Uma vez terminada a execução da música no canal 2, o valor de PLAY (2) passa a ser 0 e o MSX passa à linha seguinte do programa.

Se o valor de C for 0 o MSX testará a execução em todos os 3 canais.

## APÊNDICE D

### BIBLIOGRAFIA RECOMENDADA

A seguir apresentamos uma rápida descrição do conteúdo de alguns livros publicados pela ALEPH que poderão ser úteis aos leitores que querem se aprofundar no estudo da síntese de sons pelos micros MSX.

#### CURSO DE MÚSICA vol. 2 - Barbieri e Piazzzi

Este livro é uma sequência deste primeiro volume, abordando escalas e tonalidades; intervalos e tríades; a evolução da música tonal; os sons e o ouvido e a física da música.

#### CURSO DE BASIC MSX vol. 1 - Piazzzi e Carvalho Jr.

Este é um curso de BASIC, passo a passo, onde conceitos básicos e até alguns mais complexos são ensinados de forma suave e didática, com exemplos e exercícios (respostas no final).

#### APROFUNDANDO-SE NO MSX - Piazzzi, Maldonado, Oliveira et al

Este é um livro para quem quer entender como funciona o PSG (o gerador de sons do MSX) a nível de lógica, circuitos eletrônicos e registradores.

#### 100 DICAS PARA MSX - Oliveira et al

Neste livro podem ser encontradas 11 dicas prontas para serem usadas sobre a produção de sons nos micros MSX.

Para os leitores que querem se aprofundar no mundo da música, relacionamos a seguir alguns livros que podem ser consultados (lidos ou "tocados").

#### UMA BREVE HISTÓRIA DA MÚSICA - Roy Bennett Editora Zahar

Uma descrição sucinta da história da música desde o século IX até nossos dias.

#### A ARTE DA MÚSICA Editora Abril

Abertura da obra "MESTRES DA MÚSICA". Contém uma cronologia dos grandes compositores, uma rápida descrição da estrutura de uma orquestra sinfônica e um vocabulário de termos musicais.

**O LIVRO DA MÚSICA - Keith Spence**

Círculo do Livro

Uma abordagem bastante ampla de todo o universo musical. Fartamente ilustrado e com texto bastante atrativo.

**CURSO DE FÍSICA - ONDAS, SOM E LUZ - Ference, Lemon e Stephenson**

Editora Edgard Blucher

Este é um livro para aqueles que querem entender a música mais profundamente, do ponto de vista físico. O capítulo 3 é inteiramente dedicado à ciência da música e o capítulo 2 aborda a produção e análise dos sons, de uma maneira geral.

**LIÇÕES ELEMENTARES DE TEORIA MUSICAL - Samuel Archanjó**  
Musicália S.A.

Um curso de técnicas musicais destinado ao adestramento de músicos.

Recomendamos também que o leitor procure adquirir livros de partituras para poder treinar a transcrição para a linguagem cifrada do MSX (e apreciar a execução). Inicialmente sugerimos que se use livros para flautas (por exemplo, Minha Doce Flauta Doce, de Mário Mascarenhas, Editora Irmãos Vitale), pois as partituras são mais fáceis de serem transcritas para o MSX pois cada voz implica na emissão de uma única nota de cada vez. Para os que quiserem um trabalho um pouco mais árduo, mas não menos gratificante, aconselhamos livros de partituras de pianos (por exemplo, Invenções a Três Vozes para piano, de J.S.Bach, anotado por Souza Lima, Editora Irmãos Vitale).



No verso desta página há uma tabela que será constantemente consultada para fazer a transcrição de uma partitura para seu MSX.

Se quiser, destaque-a com cuidado e atixe-a num local visível perto do teclado do MSX.

Com o tempo você irá adquirindo familiaridade com a notação musical e ela se tornará desnecessária. Dependendo do seu grau de dedicação, depois de um certo tempo você será capaz, inclusive, de olhar uma partitura musical e cantarolar a melodia !

---

Para receber gratuitamente o Boletim Informativo da ALEPH contendo dicas, artigos e todas as informações sobre outras publicações para seu MSX, envie seu nome, endereço, CEP e o tipo de micro que você possui para:

EDITORA ALEPH  
CAIXA POSTAL 20.707  
CEP 01498 S.Paulo SP





**série didática**

# MSX


A "Série Didática" é mais uma iniciativa pioneira da EDITORA ALEPH visando levar aos seus leitores conceitos específicos através de textos claros e, realmente, didáticos. O uso do microcomputador na educação pode ser desastroso ou extremamente proveitoso, dependendo da forma como ele for usado. Na "Série Didática" a utilização dos micros é o tema central, sendo repleta de exemplos de usos proveitosos.

# curso de música

Não tem sentido tentar se aprender teoria musical sem se tocar um instrumento. A maioria das pessoas, porém, por não dispor de tempo e paciência suficientes para adquirir uma razoável habilidade num instrumento qualquer, marginaliza-se do maravilhoso mundo da música. Neste caso limitam-se, se muito, ao papel de passivos ouvintes.

Com o advento do MSX, porém, estes ouvintes passivos passam a dispor de um eficientíssimo instrumento musical de 3 vozes que vai lhes permitir aprender música interagindo (como num instrumento musical tradicional) sem que haja necessidade de um adestramento psicomotor.

O objetivo principal deste livro não é ensino de programação, apesar disso acabar ocorrendo de forma suave e didática, mas sim o aprendizado, por parte do leitor, da teoria musical em si, transformando-o num ouvinte ativo e, em certos casos, até num compositor de talento!



ALEPH PUBLICAÇÕES E ASSESSORIA PEDAGÓGICA LTDA.  
CP. 20.707 - CEP. 01498 - SÃO PAULO - (011) 843-3202